
1) ESPECIFICAÇÃO DO PROGRAMA:

1.1) Objetivos:

Implementar um corretor formal de programas utililizando o cálculo de Hoare na linguagem Prolog. Ele é interativo com o usuário, perguntando, além das pré e pós-condições do trecho de programa a ser corrigido, quais são os invariantes para os loops, já que determinar o melhor invariante não é decidível.

Este programa será desenvolvido para auxiliar os alunos de graduação e de pós-graduação da cadeira de Lógica e Especificação (INF 1632 e INF 2217, respectivamente). Com o mesmo, poderão testar exemplos vistos na teoria e ver se o programa gera uma solução igual ou parecida com a prova que fizeram de forma que os permita verificar a corretude da mesma. Ainda, podem usar o corretor de programas para ver se um dado trecho de código satisfaz um par de pré e pós-condição.

Por fim, este projeto também tem como finalidade ajudar no entendimento do funcionamento de um programa que analisa um código e determina que sentenças devem ser demonstradas para que o mesmo satisfaça determinadas propriedades. Afinal, a pesquisa a ser feita para a dissertação de mestrado é sobre Proof-Carrying Code (PCC), a qual é uma técnica que pode ser utilizada para a execução segura de código não-confiável. 

Em uma instância típica de PCC, quem vai executar o código estabelece um conjunto de regras de segurança que garantem o comportamento seguro de programas, e o produtor do código cria uma prova formal de segurança que garante a aderência às regras de segurança ao código não-confiável. Então, o primeiro é capaz de usar um validador de provas simples e rápido para verificar, com exatidão, que a prova é válida e, conseqüentemente, que o código não-confiável é seguro para ser executado. 

Na dissertação, pretende-se pesquisar como o cálculo de Hoare pode ser útil para realizar a prova formal de segurança para garantir que um código não-confiável é seguro de ser executado. Para isso, quem vai executar o código entra com as pré e pós-condições necessárias para a correção formal de programas utilizando o método citado. Já os invariantes de loops são dados pelo programador, porque é necessário ler o código para descobri-los.

1.2) Requisitos:

Os requisitos definidos para o Corretor de Programas são os seguintes:

1- O trecho de código a ser corrigido deve estar escrito na linguagem utilizada pelo corretor de programas;
2- Se houver algum erro léxico ou sintático no trecho de código a ser corrigido, o usuário deve ser avisado sobre que problema ocorreu;
3- Se houver algum erro durante operações com arquivos, o usuário deve ser avisado sobre isso e sobre que tipo de erro aconteceu;
4- As pré e pós-condições e os invariantes devem ser dados em formato correto para que possam ser aceitos. Caso contrário, devem ser pedidos novamente ao usuário;
5- Para cada loop do trecho de código a ser corrigido, um invariante deve ser pedido ao usuário;
6- O programa deve dar como resposta várias provas de correção de programa, onde para cada uma delas vai haver um conjunto de sentenças em lógica clássica que devem ser demonstradas para que a prova seja correta;
7- Para trechos de código que possuem a pós-condição derivada de sua pré-condição, deve haver pelo menos uma prova com todas as sentenças de seu conjunto demonstráveis;
8- Para trechos de código que não possuem a pós-condição derivada de sua pré-condição, não deve haver nenhuma prova com todas as sentenças de seu conjunto demonstráveis;
9- A saída das várias provas deve ser num arquivo, num formato que “facilite” o entendimento das provas pelo usuário;
10- O programa não deve entrar em loop infinito em nenhuma situação;
11- O programa deve ser facilmente mantido; e
12- Se a linguagem do trecho de código for modificada, deve ser fácil alterar o programa (seja nas análises léxica e sintática do arquivo que contém o código que será corrigido ou seja na hora de aplicar as regras do cálculo de Hoare por haver estruturas de controle diferentes). 
1.3) Fomato da Interface: 


A interface do programa é a própria interface do SWI-Prolog, se for este o interpretador Prolog sendo usado. No caso do SWI-Prolog para Windows, onde o programa foi desenvolvido e testado, foi a própria janela Windows do SWI-Prolog a interface do programa, a qual é toda textual. 

Nesta interface ocorre quase toda a interação com o usuário. É nela que são exibidas mensagens de erro ao usuário outras informações, e onde o usuário entra as informações que o programa precisa para executar. As únicas exceções são o trecho de código que o usuário passa ao programa e as provas e seus conjuntos de sentenças, as quais devem ser demonstradas para que as suas provas estejam corretas, dados como saída, o que são feitos via arquivo.

Eis aqui um exemplo da execução completa do programa na janela do SWI-Prolog para Windows quando não há erros de qualquer espécie, logo após o programa ter sido carregado na memória do interpretador da linguagem:

1.4) Funcionamento do Programa:

Nesta seção, será descrito o funcionamento do programa em linhas gerais:

1- Como o programa foi feito em Prolog, a qual é uma linguagem interpretada, ele precisa ser carregado pelo interpretador da linguagem para que o mesmo possa começar;

2- Feito isso, o usuário deve chamar o predicado do programa principal, escrevendo o nome do mesmo no prompt do interpretador;

3- Agora, o usuário deve digitar o nome do arquivo com o trecho de código a ser corrigido. Se houver um erro de abertura de arquivo, este será avisado ao usuário e o programa será abortado;

4- É feita a análise léxica do trecho de código de acordo com uma determinada linguagem. Se algum erro léxico for encontrado, o caractere inválido será exibido ao usuário. Se houver um problema na leitura do arquivo, este será avisado ao usuário. Em ambos os casos, o programa será abortado;

5- O arquivo com o trecho de código deve ser fechado após a análise léxica. Se isso não for possível, isso é avisado ao usuário e o programa, abortado; 

6- Se a análise léxica tiver sido feita com sucesso, isso é avisado ao usuário e é iniciada a análise sintática da lista de tokens resultante da análise anterior. Se algum erro sintático for encontrado, serão exibidos ao usuário os tokens reconhecidos com sucesso e a parte que não foi lida, onde provavelmente está o erro (geralmente no início da lista de tokens não-reconhecidos). Após isso, o programa é abortado. Se o trecho de código estiver sintaticamente correto, o usuário é avisado deste fato;

7- Se nenhum dos erros ditos acima tiver acontecido, o programa vai pedir a pré e a pós-condição para o trecho de código a ser corrigido. Se o formato de alguma delas estiver incorreto, a condição é pedida novamente;

8- Para cada loop (laço while) do trecho de código a ser corrigido, será pedido um invariante. Novamente, se o formato de algum deles estiver incorreto, o invariante sendo lido é pedido novamente;

9- Agora, as regras do cálculo de Hoare são aplicadas ao trecho de código a ser corrigido e são geradas várias provas de correção para o mesmo utilizando o backtracking do Prolog;

10- Para cada prova gerada, há um conjunto de sentenças que devem ser demonstradas para que a pós-condição seja válida a partir da pré-condição depois da execução do trecho de código. Serão removidas algumas dessas provas de forma que as resultantes possuam seus conjuntos de sentenças distintos entre si; e

11- As provas resultantes e seus conjuntos de sentenças são colocados em um arquivo HTML chamado saida.html. Se durante a abertura, escrita ou fechamento deste ocorrer um erro na manipulação do arquivo, uma mensagem de erro é exibida dizendo qual problema aconteceu. Caso contrário, a página HTML é gerada e o programa, terminado.
2) PROJETO MODULAR DO PROGRAMA:

2.1) Divisão Modular do Programa:

Foram feitos 7 módulos, cada um em um arquivo .pl diferente. Os módulos foram feitos de tal maneira que a funcionalidade do aplicativo ficasse bem dividida. 


Aqui há uma breve descrição de cada arquivo que compõe o projeto: 

	condicoes.pl
	-
	implementação do tratamento às pré e pós-condições, invariantes e sentenças a ser demonstradas

	erro.pl
	-
	implementação do tratamento de erros na manipulação de arquivos

	expressoes.pl
	-
	implementação da conversão de uma expressão lida pelo analisador sintático para o formato de condição

	hoare.pl
	-
	implementação das regras do cálculo de Hoare e do tratamento às provas resultantes

	lexica.pl
	-
	implementação da análise léxica

	principal.pl
	-
	módulo principal, o qual carrega os demais módulos e possui apenas o predicado que executa todo o programa 

	sintatica.pl
	-
	implementação da análise sintática


2.2) Critério de Projeto Utilizado:


Como o programa deve ler um trecho de código em uma determinada linguagem, há a necessidade de fazer as análises léxica e sintática do mesmo. Portanto, foram criados dois módulos, cada um para conter os predicados usado em cada análise.


Também, desde o início sabia-se que seria necessário um tratamento às pre e pós-condições, invariantes e sentenças; onde todos eles possuem tratamento semelhante. Portanto, há um módulo que implementa todos os predicados necessários para tratá-los.


Os outros módulos que, desde o início, já se sabia que deveriam existir são o que aplica as regras do cálculo de Hoare e aquele que possui o predicado principal do programa.


Ainda, durante a codificação, descubriu-se a necessidade de converter expressões no trecho de código para o formato de condições, o que foi posto em módulo separado por se tratar de tratamento de expressões e não de condições. 

Por fim, foi criado o módulo que trata erro em operações com arquivos. Inicialmente o predicado contido nele iria ficar no módulo da análise léxica, pois só iria tratar erros durante a leitura dos tokens do código a ser corrigido. Porém, como as provas resultantes foram colocadas em um arquivo HTML, que deve ser aberto, escrito e fechado (ou seja, que possuem operações onde podem ocorrer erros), o predicado que trata erro em operações com arquivos não mais seria de exclusividade da análise léxica e por isso foi colocado em um módulo separado.   

3) CÓDIGO FONTE:
%===============================================================================

%NOME: lexica.pl

%

%Projeto: Correção de Programas

%

%   Versão: V 1.00

%   Autora : Juliana Carpes Imperial - 0124811-7

%

%Descrição do módulo

%   Objetivo:

%      - implementação dos predicados que fazem a análise léxica da linguagem 

%        ALP.

%===============================================================================

%===============================================================================

%Nome: fazerAnaLexica

%Descrição do predicado: predicado principal do módulo, o qual realiza todas as

%                        chamadas necessárias para realizar a análise léxica.

%

%Variável de saída: 

%   - Tokens: a lista de tokens do programa em ALP.

%

%Assertiva de saída:

%   - o arquivo do programa em ALP lido está fechado.

%===============================================================================

% lê o nome de um arquivo e tenta abri-lo. Se der erro, este é tratado; se o 

% arquivo for aberto com sucesso, os tokens do arquivo ALP são lidos e se a 

% leitura dos tokens foi feita com sucesso, tenta fechar o arquivo. Se der erro,

% este é tratado, se não, exibe os tokens lidos ao usuário.

fazerAnaLexica( Tokens )                                                   :- 

   write( '\nEscaneando: análise léxica...\n' )                            , 

   write( 'Entre nome do arquivo fonte entre \' \' e seguido de ponto: ' ) ,  

   read( Arquivo )                                                         , 

   catch( see( Arquivo ), _Excecao, tratarErro( abertura ) )               ,

   nl                                                                      , 

   escanear( Tokens )                                                      ,  

   catch( seen, _Excecao, tratarErro( fechamento ) )                       ,

   write( '\nAnálise léxica feita com sucesso.\n' )                        , 

   write( '\ntokens = ' )                                                  ,

   write( Tokens )                                                         , 

   nl                                                                      ,

   !                                                                       .

%===============================================================================

%Nome: serEspaco

%Descrição do predicado: determina se um caractere lido é o espaço, ou seja, se 

%                        seu código ASCII é 32.

%

%Variável de entrada: 

%   - o código ASCII do caractere lido.

%===============================================================================

serEspaco( 32 ) .     

%===============================================================================

%Nome: serTab

%Descrição do predicado: determina se um caractere lido é o tab, ou seja, se seu 

%                        código ASCII é 9.

%

%Variável de entrada: 

%   - o código ASCII do caractere lido.

%===============================================================================

serTab( 9 ) .          

%===============================================================================

%Nome: serFimLinha

%Descrição do predicado: determina se um caractere lido é o que determina fim de 

%                        linha, ou seja, se seu código ASCII é 10. 

%

%Variável de entrada: 

%   - o código ASCII do caractere lido.

%===============================================================================

serFimLinha( 10 ) .   

%===============================================================================

%Nome: serFimArq

%Descrição do predicado: determina se um caractere lido é o que determina fim de 

%                        arquivo, ou seja, se seu código ASCII é 26 ou -1.

%

%Variável de entrada: 

%   - o código ASCII do caractere lido. 

%===============================================================================

serFimArq( 26 ) .   

serFimArq( -1 ) .

%===============================================================================

%Nome: serPonto

%Descrição do predicado: determina se um caractere lido é o ponto ("."), ou 

%                        seja, se seu código ASCII é 46.

%

%Variável de entrada: 

%   - o código ASCII do caractere lido.

%===============================================================================

serPonto( 46 ) .

%===============================================================================

%Nome: serSimples

%Descrição do predicado: determina se um caractere lido é tal que, se for o 

%                        primeiro de um token, é o próprio token.

%

%Variável de entrada: 

%   - o código ASCII do caractere lido.

%

%Variável de saída:

%   - o nome do token correspondente ao caractere lido.     

%===============================================================================

serSimples( 40, parenEsq ) .   % caractere: ( -- token: parenEsq

serSimples( 41, parenDir ) .   % caractere: ) -- token: parenDir

serSimples( 42, mult )     .   % caractere: * -- token: mult

serSimples( 43, soma )     .   % caractere: + -- token: soma

serSimples( 45, menos )    .   % caractere: - -- token: menos

serSimples( 47, divide )   .   % caractere: / -- token: divide

serSimples( 59, ptEVirg )  .   % caractere: ; -- token: ptEVirg

serSimples( 61, igual )    .   % caractere: = -- token: igual

%===============================================================================

%Nome: serDuplo

%Descrição do predicado: determina se um caractere lido é tal que, se for o 

%                        primeiro de um token, ou é o próprio token ou é o 

%                        primeiro de um token com dois caracteres.

%

%Variável de entrada: 

%   - o código ASCII do caractere lido.

%

%Variável de saída:

%   - o nome do token correspondente ao caractere lido se o primeiro só contiver

%     um caractere.     

%===============================================================================

serDuplo( 58, doisPts ) .   % caractere: : -- token: doisPts

serDuplo( 60, menor )   .   % caractere: < -- token: menor

serDuplo( 62, maior )   .   % caractere: > -- token: maior

%===============================================================================

%Nome: serPar

%Descrição do predicado: determina se dois caracteres lidos são tais que, se 

%                        forem lidos em seqüência, formam um token de dois

%                        caracteres.

%

%Variáveis de entrada: 

%   - o código ASCII do primeiro caractere lido; e

%   - o código ASCII do segundo caractere lido.

%

%Variável de saída:

%   - o nome do token correspondente ao par de caracteres lidos.

%===============================================================================

serPar( 58, 61, atribui ) .   % par de caracteres: := -- token: atribui

serPar( 60, 61, menorIg ) .   % par de caracteres: <= -- token: menorIg

serPar( 62, 61, maiorIg ) .   % par de caracteres: >= -- token: maiorIg

serPar( 60, 62, difer )   .   % par de caracteres: <> -- token: difer

%===============================================================================

%Nome: serReserv

%Descrição do predicado: determina se um identificador corresponde a uma palavra 

%                        reservada da linguagem ALP.                         

%

%Variável de entrada: 

%   - o nome do identificador. 

%===============================================================================

serReserv( skip )    .   % palavra reservada skip (comando vazio)

serReserv( if )      .   % palavra reservada if (teste)  

serReserv( then )    .   % palavra reservada then (caso positivo do teste) 

serReserv( fi )      .   % palavra reservada fi (fim de if-then)

serReserv( else )    .   % palavra reservada else (caso negativo do teste)

serReserv( esle_fi ) .   % palavra reservada esle_fi (fim de if-then-else)

serReserv( while )   .   % palavra reservada while (repetição)

serReserv( do )      .   % palavra reservada do (o que faz a repetição)

serReserv( od )      .   % palavra reservada od (fim de while)

serReserv( and )     .   % palavra reservada and (conectivo booleano)

serReserv( or )      .   % palavra reservada or (conectivo booleano)

serReserv( not )     .   % palavra reservada not (conectivo booleano) 

serReserv( true )    .   % palavra reservada true (booleano primitivo)

serReserv( false )   .   % palavra reservada false (booleano primitivo) 

%===============================================================================

%Nome: serBranco

%Descrição do predicado: determina se um caractere lido é um branco.

%

%Variável de entrada: 

%   - Car: o código ASCII do caractere lido.

%===============================================================================

serBranco( Car )      :-  % um caractere é um branco se:

   serEspaco( Car )   ;   % é um espaço, ou

   serTab( Car )      ;   % é um tab, ou

   serFimLinha( Car ) .   % é um fim de linha

%===============================================================================

%Nome: serMinuscula

%Descrição do predicado: determina se um caractere lido corresponde a uma letra

%                        minúscula.

%

%Variável de entrada: 

%   - Car: o código ASCII do caractere lido.

%===============================================================================

serMinuscula( Car ) :-  % um caractere é uma letra minúscula se:

   Car >= 97        ,   % é 'a' (ASCII = 97) ou vem depois dele e

   Car =< 122       .   % é 'z' (ASCII = 122) ou vem antes dele    

%===============================================================================

%Nome: serMaiuscula

%Descrição do predicado: determina se um caractere lido corresponde a uma letra

%                        maiúscula.

%

%Variável de entrada: 

%   - Car: o código ASCII do caractere lido.

%===============================================================================

serMaiuscula( Car ) :-  % um caractere é uma letra maiúscula se:

   Car >= 65        ,   % é 'A' (ASCII = 65) ou vem depois dele e

   Car =< 90        .   % é 'Z' (ASCII = 90) ou vem antes dele        

%===============================================================================

%Nome: serDigito 

%Descrição do predicado: determina se um caractere lido corresponde a um 

%                        dígito decimal.

%

%Variável de entrada: 

%   - Car: o código ASCII do caractere lido.

%===============================================================================

serDigito( Car ) :-  % um caractere é um dígito decimal se:        

   Car >= 48     ,   % é '0' (ASCII = 48) ou vem depois dele e

   Car =< 57     .   % é '9' (ASCII = 57) ou vem antes dele    

%===============================================================================

%Nome: serCarVal

%Descrição do predicado: determina se um caractere lido é válido dentro de um 

%                        identificador.

%

%Variável de entrada: 

%   - Car: o código ASCII do caractere lido.

%===============================================================================

serCarVal( Car )       :-  % um caractere é válido dentro de um identificador se                    

   serMinuscula( Car ) ;   % é uma letra minúscula, ou

   serMaiuscula( Car ) ;   % é uma letra maiúscula, ou

   serDigito( Car )    ;   % é um dígito decimal, ou

   Car == 95           .   % é um sublinhado ('_', ASCII = 95)

%===============================================================================

%Nome: escanear

%Descrição do predicado: escanea a lista de caracteres e monta a lista de tokens

%                        do programa em ALP sendo lido.

%

%Variável de saída: 

%   - ListaTokens: a lista de tokens do programa em ALP lido.

%

%Assertiva de entrada:

%   - o arquivo com o programa ALP está aberto.

%===============================================================================

escanear( [ Token | ListaTokens ] )         :-

   tab( 4 )                                 ,   % imprime 4 brancos

   pegarCar( Car )                          ,   % pega o primeiro caractere

   pegarToken( Car, Token, NovoCar )        ,   % pega o primeiro token 

   restoProg( Token, NovoCar, ListaTokens ) ,   % escanea o resto dos tokens

   !                                        .

%===============================================================================

%Nome: pegarCar

%Descrição do predicado: pega o próximo caractere do código fonte do programa em

%                        ALP sendo lido.

%

%Variável de saída: 

%   - Car: um caractere presente no código fonte do programa em ALP.

%

%Assertiva de entrada:

%   - o arquivo com o programa ALP está aberto.

%===============================================================================

% lê um caractere, se houver erro, este é tratado; se não der erro e ele for fim 

% de linha, uma linha é pulada e 4 brancos são impressos; já se ele for fim de 

% arquivo, uma linha é pulada; por fim, se não for fim de linha ou de arquivo, o 

% imprime

pegarCar( Car )                                          :- 

   catch( get0( Car ), _Excecao, tratarErro( leitura ) ) ,  

   ( serFimLinha( Car )                                  ,   

     nl                                                  ,    

     tab( 4 )                                            ;    

     serFimArq( Car )                                    ,   

     nl                                                  ;  

     put( Car ) )                                        .   

%===============================================================================

%Nome: restoProg

%Descrição do predicado: escanea o restante da lista de caracteres e monta o 

%                        restante da lista de tokens do programa em ALP sendo 

%                        lido.

%

%Variáveis de entrada: 

%   - UltToken: o último token lido; e

%   - Car:      o último caractere lido.

%

%Variável de saída:

%   - ListaTokens: a lista de tokens do programa em ALP lido posteriores a

%                  UltToken.

%

%Assertiva de entrada:

%   - o arquivo com o programa ALP está aberto.

%===============================================================================

% se ler depois do fim de arquivo, não retorna tokens

restoProg( eop, _Car, [ ] ) .   

% se não ler depois do fim de arquivo, pega o próximo token, o coloca na lista

% de tokens e pega o restante da lista de tokens recursivamente

restoProg( _UltToken, Car, [ Token | ListaTokens ] ) :- 

   pegarToken( Car, Token, NovoCar )                 ,   

   restoProg( Token, NovoCar, ListaTokens )          .

%===============================================================================

%Nome: pegarToken

%Descrição do predicado: pega o próximo token do programa em ALP sendo lido.

%

%Variável de entrada: 

%   - Car: o último caractere lido.

%

%Variáveis de saída:

%   - Token:     o token lido; e

%   - NovoCar_2: o caractere lido mais recentemente.

%

%Assertiva de entrada:

%   - o arquivo com o programa ALP está aberto.

%

%Assertiva de saída: 

%   - o arquivo manipulado continuará aberto a menos que um caractere inválido 

%     seja lido, o que fará com que o programa seja abortado.

%===============================================================================

% se ler fim de arquivo, retorna o token correspondente 

pegarToken( Car, eop, 0 ) :-  

   serFimArq( Car )       .

% pega tokens com apenas um caractere 

pegarToken( Car, Token, NovoCar_2 ) :-

   serSimples( Car, Token )         ,

   pegarCar( NovoCar_2 )            .

% pega token que pode ser tanto simples ou duplo:

% se o primeiro caractere lido não fizer par com o próximo, o token é simples

% caso contrário, é duplo

pegarToken( Car, Token, NovoCar_2 )  :-  

   serDuplo( Car, OutroToken )       ,

   pegarCar( NovoCar_1 )             ,

   ( serPar( Car, NovoCar_1, Token ) ,

     pegarCar( NovoCar_2 )           ; 

     Token = OutroToken              ,

     NovoCar_2 = NovoCar_1 )         .

% pega token que é um identificador, que tem como primeiro caractere uma letra

% o resto do token pode ser composto de letras, dígitos e sublinhados

% se o identificador for palavra reservada de ALP, apenas ele é retornado

% senão, ele é retornardo como um predicado que diz que ele é identificador

pegarToken( Car, Token, NovoCar_2 )               :-  

   ( serMinuscula( Car )                          ;

     serMaiuscula( Car ) )                        , 

   pegarCar( NovoCar_1 )                          , 

   pegarRestoId( NovoCar_1, ListaCar, NovoCar_2 ) ,   

   name( Id, [ Car | ListaCar ] )                 , 

   ( serReserv( Id )                              ,

     Token = Id                                   ; 

     Token = ide( Id ) )                          .

% pega token que é um número, o qual é composto apenas de dígitos

pegarToken( Car, num( Num ), NovoCar_2 )           :- 

   serDigito( Car )                                ,

   pegarCar( NovoCar_1 )                           ,

   pegarRestoNum( NovoCar_1, ListaCar, NovoCar_2 ) ,

   name( Num, [ Car | ListaCar ] )                 .

% se o último caractere lido é um branco, apenas lê um novo caractere

% e pega um token a partir dele

pegarToken( Car, Token, NovoCar_2 )        :-

   serBranco( Car )                        , 

   pegarCar( NovoCar )                     , 

   pegarToken( NovoCar, Token, NovoCar_2 ) .

% se o último caractere lido é inválido, ou seja, não pode pertencer a nenhum 

% token, isto é avisado ao usuário, e o programa é abortado, fechando antes o

% arquivo aberto

pegarToken( Car, _Token, _NovoCar_2 )  :-

   write( '\n\nCaractere ilegal: ' )   ,

   put( Car )                          ,

   nl                                  ,

   catch( seen, _Excecao, abort )      ,

   abort                               .

%===============================================================================

%Nome: pegarRestoId

%Descrição do predicado: pega o restante de um identificador.

%

%Variável de entrada: 

%   - Car: o último caractere lido.

%

%Variáveis de saída:

%   - ListaCar:  a lista de caracteres lidos que representa um identificador; e

%   - NovoCar_2: o caractere lido mais recentemente.

%

%Assertiva de entrada:

%   - o arquivo com o programa ALP está aberto.

%===============================================================================

% se o último caractere lido puder ser encontrado dentro de um identificador, a 

% busca por caracteres que possam pertencer a ele continua recursivamente

pegarRestoId( Car, [ Car | ListaCar ], NovoCar_2 ) :- 

   serCarVal( Car )                                , 

   pegarCar( NovoCar_1 )                           ,

   pegarRestoId( NovoCar_1, ListaCar, NovoCar_2 )  .

% se o último caractere lido não puder ser encontrado dentro de um 

% identificador, pára a busca

pegarRestoId( Car, [ ], Car ) .

%===============================================================================

%Nome: pegarRestoNum

%Descrição do predicado: pega o restante de um número.

%

%Variável de entrada: 

%   - Car: o último caractere lido.

%

%Variáveis de saída:

%   - ListaCar:  a lista de caracteres lidos que representa um número; e

%   - NovoCar_2: o caractere lido mais recentemente.

%

%Assertiva de entrada:

%   - o arquivo com o programa ALP está aberto.

%===============================================================================

% se o último caractere lido for um dígito, a busca por dígitos continua 

% recursivamente

pegarRestoNum( Car, [ Car | ListaCar ], NovoCar_2 ) :- 

   serDigito( Car )                                 , 

   pegarCar( NovoCar_1 )                            , 

   pegarRestoNum( NovoCar_1, ListaCar, NovoCar_2 )  .

% se o último caractere lido não for um dígito, pára a busca

pegarRestoNum( Car, [ ], Car ) .

%===============================================================================

%NOME: sintatica.pl

%

%Projeto: Correção de Programas

%

%   Versão: V 1.00

%   Autora : Juliana Carpes Imperial - 0124811-7

%

%Descrição do módulo

%   Objetivos:

%      - implementação das clásulas que fazem a análise sintática da linguagem 

%        ALP, utilizando regras de DCG (Definite Clause Grammars); e

%      - implementação de um predicado que imprime o primeiro erro de sintaxe

%        encontrado, se for o caso. 

%===============================================================================

%===============================================================================

%Nome: fazerAnaSintatica

%Descrição do predicado: predicado principal do módulo, o qual realiza todas as

%                        chamadas necessárias para realizar a análise sintática.

%

%Variável de entrada:

%   - Tokens: a lista de tokens do código fonte do programa em ALP.

%

%Variável de saída:

%   - Arvore: a árvore sintática do programa em ALP.

%===============================================================================

% analisa os tokens lidos do código-fonte e monta a árvore sintática 

% correspondente se o programa estiver sintáticamente correto (será abortado se

% não o estiver)

fazerAnaSintatica( Tokens, Arvore )                   :-

   write( '\nInterpretando: análise sintática...\n' ) ,

   serComandos( Arvore, Tokens, Tokens, [ eop ] )     ,

   write( 'Análise sintática feita com sucesso.\n' )  ,

   write( '\narvore sintática = ' )                   ,

   write( Arvore )                                    ,

   nl                                                 ,

   !                                                  .

%===============================================================================

%Nome: serComandos

%Descrição da cláusula: reconhece a lista de comandos do programa em ALP.

%

%Variável de entrada: 

%   - Tokens: a lista de tokens do código fonte.

%

%Variável de saída:

%   - Comandos: a lista de comandos dados no código fonte se não houve erro

%               sintático. Se houve, será chamada a rotina de tratamento de 

%               erro e o programa, abortado.

%

%Cláusula na BNF de ALP:

%   - < cmds > ::= < cmd > | < cmd > ; < cmds >

%===============================================================================

serComandos( Comandos, Tokens )              -->                                     

   serComando( Comando, Tokens )             ,   % reconhece primeiro comando

   serRestoCmds( Comando, Comandos, Tokens ) .   % reconhece os demais comandos

% se tentou esta cláusula, houve erro sintático, portanto, o erro será tratado  

serComandos( _Comandos, Tokens ) -->

   tratarErro( Tokens )          .

%===============================================================================

%Nome: serRestoCmds

%Descrição da cláusula: reconhece o restante da lista de comandos do programa em 

%                       ALP.

%

%Variáveis de entrada: 

%   - Tokens:  a lista de tokens do código fonte; e

%   - Comando: o último comando reconhecido.

%

%Variável de saída:

%   - Comandos: a lista de comandos reconhecidos, incluindo o último comando, se

%               não houve erro sintático; se houve, o programa será abortado.

%===============================================================================

% reconhece o restante da lista de comandos, reconhecendo, antes, o ";" que

% veio depois de um comando já reconhecido

serRestoCmds( Comando, [ Comando | Comandos ], Tokens ) -->

   [ ptEVirg ]                                          ,    

   serComandos( Comandos, Tokens )                      .     

% não há mais comandos a serem reconhecidos; todos já o foram 

serRestoCmds( Comando, [ Comando ], _Tokens ) -->

   [ ]                                        . 

%===============================================================================

%Nome: serComando

%Descrição da cláusula: reconhece um comando do programa em ALP.

%

%Variável de entrada: 

%   - Tokens: a lista de tokens do código fonte.

%

%Variável de saída:

%   - Comando: uma árvore com a estrutura do comando se não houve erro

%              sintático. Se houve, será chamada a rotina de tratamento de erro

%              e programa, abortado.

%

%Cláusula na BNF de ALP:

%   - < cmd > ::= < var > := < exp >

%               | skip

%               | if < bexp > then < cmds > fi

%               | if < bexp > then < cmds > else < cmds > esle_fi

%               | while < bexp > do < cmds > od

%===============================================================================

% reconhece um comando de atribuição, reconhecendo primeiro a variável e, 

% posteriormente, a expressão que é atribuída a primeira

serComando( atribui( Variavel, Expressao ), Tokens ) -->

   [ ide( Variavel ), atribui ]                      ,   

   serExpressao( Expressao, Tokens )                 .    

% reconhece um comando skip

serComando( skip, _Tokens ) -->

   [ skip ]                 .

% reconhece um comando if, reconhecendo, nesta ordem, o identificador if, uma 

% expressão booleana, o identificador then, os comandos que serão executados

% caso o teste do if seja bem sucedido e, por fim, o resto do comando if

% (lê a parte do else se ela existir)

serComando( Comando, Tokens )                  -->

   [ if ]                                      ,   

   serBoolExpr( Teste, Tokens )                ,   

   [ then ]                                    ,  

   serComandos( Entao, Tokens )                ,

   serRestoIf( Teste, Entao, Comando, Tokens ) .

% reconhece um comando while, reconhecendo, nesta ordem, o identificador 

% while, uma expressão booleana, o identificador do, os comandos do corpo

% do while e o identificador od, indicando fim do while

serComando( while( Teste, Comandos ), Tokens ) -->

   [ while ]                                   ,

   serBoolExpr( Teste, Tokens )                ,

   [ do ]                                      ,

   serComandos( Comandos, Tokens )             ,

   [ od ]                                      . 

% se tentou esta cláusula, houve erro sintático, portanto, o erro será tratado  

serComando( _Comando, Tokens ) -->

   tratarErro( Tokens )        .

%===============================================================================

%Nome: serRestoIf

%Descrição da cláusula: reconhece o restante de um comando if.

%

%Variáveis de entrada: 

%   - Tokens: a lista de tokens do código fonte;

%   - Teste:  a expressão booleana a ser avaliada pelo comando if; e

%   - Então:  a lista de comandos executada se o teste do if for bem sucedido.

%

%Variável de saída:

%   - if( ... ): a árvore com a estrutura do comando if.

%===============================================================================

% se for um if-then-else, reconhece, nesta ordem, o identificador else, os 

% comandos que serão executados caso o teste do if não seja bem sucedido e o 

% e o identificador esle_fi, indicando fim do if-then-else

serRestoIf( Teste, Entao, if( Teste, Entao, Senao ), Tokens ) -->

   [ else ]                                                   , 

   serComandos( Senao, Tokens )                               ,

   [ esle_fi ]                                                .

% se for um if-then, reconhece apenas o identificador fi, indicando fim do 

% if-then, já que todo o comando if já foi reconhecido

serRestoIf( Teste, Entao, if( Teste, Entao ), _Tokens ) -->

   [ fi ]                                               .

%===============================================================================

%Nome: serExpressao

%Descrição da cláusula: reconhece uma expressão numérica ou booleana do programa 

%                       em ALP.

%

%Variável de entrada: 

%   - Tokens: a lista de tokens do código fonte.

%

%Variável de saída:

%   - Expressão: uma árvore com a estrutura da expressão se não houve erro 

%                sintático. Se houve, será chamada a rotina de tratamento de 

%                erro e o programa, abortado.

%

%Cláusula na BNF de ALP:

%   - < exp > ::= < bexp > | < iexp >

%===============================================================================

serExpressao( Expressao, Tokens )   --> 

   serIntExpr( Expressao )          ;   % reconhece expressão numérica, ou

   serBoolExpr( Expressao, Tokens ) .   % reconhece expressão booleana

% se tentou esta cláusula, houve erro sintático, portanto, o erro será tratado  

serExpressao( _Expressao, Tokens ) -->

   tratarErro( Tokens )            .              

%===============================================================================

%Nome: serIntExpr

%Descrição da cláusula: reconhece uma expressão numérica do programa em ALP.

%

%Variável de saída:

%   - Expressão: uma árvore com a estrutura da expressão numérica se não houve 

%                erro sintático. Se houve, o programa será abortado.

%

%Cláusula na BNF de ALP:

%   - < iexp > ::= < iterm > | < iterm > < fraco > < iexp >

%===============================================================================

serIntExpr( Expressao )                --> 

   serIntTermo( Termo )                ,   % reconhece o primeiro termo da expr

   serRestoIntExpr( Termo, Expressao ) .   % reconhece o resto da expr

%===============================================================================

%Nome: serRestoIntExpr

%Descrição da cláusula: reconhece o resto de uma expressão numérica do programa 

%                       em ALP.

%

%Variável de entrada:

%   - Termo: o último termo reconhecido da expressão sendo reconhecida.

%

%Variável de saída:

%   - Expressão: uma árvore com a estrutura da expressão numérica se não houve 

%                erro sintático. Se houve, o programa será abortado.

%===============================================================================

% reconhece o operador da expressão (+ / -), o termo seguinte, monta um termo

% com o termo anterior, o lido desta vez e o operador e, por fim, reconhece o 

% restante da expressão

serRestoIntExpr( Termo, Expressao )                                    --> 

   serOperFraco( Operador )                                            , 

   serIntTermo( OutroTermo )                                           , 

   serRestoIntExpr( expInt( Operador, Termo, OutroTermo ), Expressao ) .

% toda a expressão já foi reconhecida; nada mais precisa ser reconhecido

serRestoIntExpr( Expressao, Expressao ) --> 

   [ ]                                  .

%===============================================================================

%Nome: serIntTermo

%Descrição da cláusula: reconhece um termo numérico do programa em ALP.

%

%Variável de saída:

%   - Termo: uma árvore com a estrutura do termo numérico se não houve erro 

%            sintático. Se houve, o programa será abortado.

%

%Cláusula na BNF de ALP:

%   - < iterm > ::= < ielem > | < ielem > < forte > < iterm >

%===============================================================================

serIntTermo( Termo )                   --> 

   serIntElem( Elemento )              ,   % reconhece o primeiro elem do termo

   serRestoIntTermo( Elemento, Termo ) .   % reconhece o resto do termo

%===============================================================================

%Nome: serRestoIntTermo

%Descrição da cláusula: reconhece o resto de um termo numérico do programa em 

%                       ALP.

%

%Variável de entrada:

%   - Elemento: o último elemento reconhecido do termo sendo reconhecido.

%

%Variável de saída:

%   - Termo: uma árvore com a estrutura do termo numérico se não houve erro 

%            sintático. Se houve, o programa será abortado.

%===============================================================================

% reconhece o operador do termo (* / /), o elemento seguinte, monta um elemento

% com o elemento anterior, o lido desta vez e o operador e, por fim, reconhece o 

% restante do termo

serRestoIntTermo( Elemento, Termo )                                   --> 

   serOperForte( Operador )                                           , 

   serIntElem( OutroElem )                                            , 

   serRestoIntTermo( expInt( Operador, Elemento, OutroElem ), Termo ) .

% todo o termo já foi reconhecido; nada mais precisa ser reconhecido

serRestoIntTermo( Termo, Termo ) --> 

   [ ]                           .

%===============================================================================

%Nome: serIntElem

%Descrição da cláusula: reconhece um elemento numérico do programa em ALP.

%

%Variável de saída:

%   - uma árvore com a estrutura do elemento numérico se não houve erro 

%     sintático. Se houve, o programa será abortado.

%

%Cláusula na BNF de ALP:

%   - < ielem > ::= < num > 

%                 | < var > 

%                 | - ( < iexp > )

%                 | ( < iexp > )

%===============================================================================

% o elemento numérico reconhecido é um número inteiro

serIntElem( num( Num ) ) --> 

   [ num( Num ) ]        .

% o elemento numérico reconhecido é uma variável

serIntElem( ide( Identificador ) ) -->

   [ ide( Identificador ) ]        .

% o elemento numérico reconhecido é um elemento com um sinal de menos na frente

serIntElem( menos( Elemento ) ) --> 

   [ menos ]                    ,   % reconhece sinal de menos 

   serIntElem( Elemento )       .   % reconhece o resto do elemento

% o elemento numérico reconhecido é uma expressão entre parênteses

serIntElem( Expressao )    --> 

   [ parenEsq ]            ,   % reconhece parêntese esquerdo

   serIntExpr( Expressao ) ,   % reconhece expressão numérica

   [ parenDir ]            .   % reconhece parêntese direito

%===============================================================================

%Nome: serBoolExpr

%Descrição da cláusula: reconhece uma expressão booleana do programa em ALP.

%

%Variável de entrada: 

%   - Tokens: a lista de tokens do código fonte.

%

%Variável de saída:

%   - Expressão: uma árvore com a estrutura da expressão booleana se não houve 

%                erro sintático. Se houve, será chamada a rotina de tratamento 

%                de erro e o programa, abortado.

%

%Cláusula na BNF de ALP:

%   - < bexp > ::= < bterm > | < bterm > or < bexp >

%===============================================================================

serBoolExpr( Expressao, Tokens )                --> 

   serBoolTermo( Termo, Tokens )                ,   % reconhece o primeiro termo

   serRestoBoolExpr( Termo, Expressao, Tokens ) .   % reconhece o resto da expr

% se tentou esta cláusula, houve erro sintático, portanto, o erro será tratado  

serBoolExpr( _Expressao, Tokens ) -->

   tratarErro( Tokens )           .

%===============================================================================

%Nome: serRestoBoolExpr

%Descrição da cláusula: reconhece o resto de uma expressão booleana do programa 

%                       em ALP.

%

%Variáveis de entrada:

%   - Tokens: a lista de tokens do código fonte; e

%   - Termo:  o último termo reconhecido da expressão sendo reconhecida.

%

%Variável de saída:

%   - Expressão: uma árvore com a estrutura da expressão booleana se não houve 

%                erro sintático. Se houve, o programa será abortado.

%===============================================================================

% reconhece o operador da expressão (or), o termo seguinte, monta um termo com o 

% termo anterior, o lido desta vez e o operador e, por fim, reconhece o restante 

% da expressão

serRestoBoolExpr( Termo, Expressao, Tokens )                               --> 

   [ or ]                                                                  , 

   serBoolTermo( OutroTermo, Tokens )                                      , 

   serRestoBoolExpr( expBool( or, Termo, OutroTermo ), Expressao, Tokens ) .

% toda a expressão já foi reconhecida; nada mais precisa ser reconhecido

serRestoBoolExpr( Expressao, Expressao, _Tokens ) --> 

   [ ]                                            .

%===============================================================================

%Nome: serBoolTermo

%Descrição da cláusula: reconhece um termo booleano do programa em ALP.

%

%Variável de entrada: 

%   - Tokens: a lista de tokens do código fonte.

%

%Variável de saída:

%   - Termo: uma árvore com a estrutura do termo booleano se não houve erro 

%            sintático. Se houve, o programa será abortado.

%

%Cláusula na BNF de ALP:

%   - < bterm > ::= < belem > | < belem > and < bterm >

%===============================================================================

serBoolTermo( Termo, Tokens )                   --> 

   serBoolElem( Elemento, Tokens )              ,   % reconhece o primeiro elem

   serRestoBoolTermo( Elemento, Termo, Tokens ) .   % reconhece o resto do termo

%===============================================================================

%Nome: serRestoBoolTermo

%Descrição da cláusula: reconhece o resto de um termo booleano do programa em 

%                       ALP.

%

%Variáveis de entrada:

%   - Tokens:   a lista de tokens do código fonte; e

%   - Elemento: o último elemento reconhecido do termo sendo reconhecido.

%

%Variável de saída:

%   - Termo: uma árvore com a estrutura do termo booleano se não houve erro 

%            sintático. Se houve, o programa será abortado.

%===============================================================================

% reconhece o operador do termo (and), o elemento seguinte, monta um elemento

% com o elemento anterior, o lido desta vez e o operador e, por fim, reconhece o 

% restante do termo

serRestoBoolTermo( Elemento, Termo, Tokens )                               --> 

   [ and ]                                                                 , 

   serBoolElem( OutroElem, Tokens )                                        , 

   serRestoBoolTermo( expBool( and, Elemento, OutroElem ), Termo, Tokens ) .

% todo o termo já foi reconhecido; nada mais precisa ser reconhecido

serRestoBoolTermo( Termo, Termo, _Token ) --> 

   [ ]                                    .

%===============================================================================

%Nome: serBoolElem

%Descrição da cláusula: reconhece um elemento booleano do programa em ALP.

%

%Variável de entrada:

%   - Tokens: a lista de tokens do código fonte.

%

%Variável de saída:

%   - uma árvore com a estrutura do elemento booleano se não houve erro

%     sintático. Se houve, o programa será abortado.

%

%Cláusulas na BNF de ALP:

%   - < belem > ::= true 

%                 | false 

%                 | < var > 

%                 | not ( < bexp > )

%                 | ( < bexp > )

%                 | < comp >

%   - < comp > ::= < iexp > < rel > < iexp > 

%===============================================================================

% o elemento booleano reconhecido é a constante true

serBoolElem( true, _Tokens ) --> 

   [ true ]                  .

% o elemento booleano reconhecido é a constante false

serBoolElem( false, _Tokens ) -->

   [ false ]                  .

% o elemento booleano reconhecido é uma variável

serBoolElem( ide( Identificador ), _Tokens ) -->

   [ ide( Identificador ) ]                  .

% o elemento booleano reconhecido é um elemento negado

serBoolElem( neg( Elemento ), Tokens ) --> 

   [ not ]                             ,   % reconhece operador not

   serBoolElem( Elemento, Tokens )     .   % reconhece o resto do elemento

% o elemento booleano reconhecido é uma expressão entre parênteses

serBoolElem( Expressao, Tokens )    --> 

   [ parenEsq ]                     ,   % reconhece parêntese esquerdo

   serBoolExpr( Expressao, Tokens ) ,   % reconhece expressão booleana

   [ parenDir ]                     .   % reconhece parêntese direito

% o elemento booleano é uma comparação entre expressões numéricas, onde uma

% expressão numérica é reconhecida, seguida de um operador de relação de ordem

% e, por fim, é reconhecida outra expressão numérica

serBoolElem( expBool( Relacao, Expressao_1, Expressao_2 ), _Tokens ) -->

   serIntExpr( Expressao_1 )                                         ,

   serRelacao( Relacao )                                             ,

   serIntExpr( Expressao_2 )                                         .

%===============================================================================

%Nome: serOperFraco

%Descrição da cláusula: reconhece um operador de adição ou subtração.

%

%Variável de saída:

%   - o operador reconhecido.

%

%Cláusula na BNF de ALP:

%   - < fraco > ::= + | -

%===============================================================================

serOperFraco( soma ) -->

   [ soma ]          .    % reconhece o operador de adição

serOperFraco( menos ) -->

   [ menos ]          .   % reconhece o operador de subtração

%===============================================================================

%Nome: serOperForte

%Descrição da cláusula: reconhece um operador de multiplicação ou divisão.

%

%Variável de saída:

%   - o operador reconhecido.

%

%Cláusula na BNF de ALP:

%   - < forte > ::= * | /

%===============================================================================

serOperForte( mult ) -->

   [ mult ]          .     % reconhece o operador de multiplicação

serOperForte( divide ) -->

   [ divide ]          .   % reconhece o operador de divisão

%===============================================================================

%Nome: serRelacao

%Descrição da cláusula: reconhece um operador de comparação entre expressões

%                       numéricas.

%

%Variável de saída:

%   - o operador de relação reconhecido.

%

%Cláusula na BNF de ALP:

%   - < rel > ::= <= | < | = | >= | <>

%===============================================================================

serRelacao( igual ) -->

   [ igual ]         .    % reconhece o operador de igualdade

serRelacao( difer ) -->

   [ difer ]         .    % reconhece o operador de desigualdade

serRelacao( menor ) -->

   [ menor ]        .     % reconhece o operador de menor do que

serRelacao( maior ) -->

   [ maior ]        .     % reconhece o operador de maior do que

serRelacao( menorIg ) -->

   [ menorIg ]        .   % reconhece o operador de menor ou igual do que

serRelacao( maiorIg ) -->

   [ maiorIg ]        .   % reconhece o operador de maior ou igual do que

%===============================================================================

%Nome: tratarErro

%Descrição do predicado: trata um erro encontrado durante a análise sintática.

%

%Variáveis de entrada:

%   - Tokens:    a lista de tokens do código fonte; e

%   - Restantes: os tokens que ainda não foram reconhecidos pelo analisador

%                sintático.

%

%Variável de saída:

%   - Inútil: não retorna nada, já que o programa é abortado durante a execução

%             deste predicado. Esta aí apenas por causa da transformação das

%             cláusulas DCG em predicados PROLOG.

%===============================================================================

% acha os tokens reconhecidos, pois é a primeira parte da lista dos tokens do

% código fonte, cuja última parte são os que não foram reconhecidos por causa do

% erro

tratarErro( Tokens, Restantes, _Inutil ) :-

   nl                                    ,   

   write( 'Tokens reconhecidos: ' )      ,

   append( Lidos, Restantes, Tokens )    ,   % acha os tokens reconhecidos

   write( Lidos )                        ,   % imprime os reconhecidos

   nl                                    ,   

   write( 'Houve erro aqui: ' )          ,   % aponta onde houve o erro

   write( Restantes )                    ,   % imprime os não reconhecidos

   nl                                    ,   

   abort                                 .   % aborta o programa

%===============================================================================

%NOME: condicoes.pl

%

%Projeto: Correção de Programas

%

%   Versão: V 1.00

%   Autora : Juliana Carpes Imperial - 0124811-7

%

%Descrição do módulo

%   Objetivos:

%      - ler as pré e pós-condições, e os invariantes que serão utilizados

%        durante a correção do programa em ALP; 

%      - determinar se as pré e pós-condições, e os invariantes estão 

%        sintaticamente corretos; 

%      - achar os loops existentes no programa escrito em ALP reconhecido, para

%        que seus invariantes possam ser dados pelo usuário;

%      - procurar o invariante de um determinado loop em uma lista de

%        < loop, invariante >;

%      - ver se uma pré-condição é equivalente a uma pós-condição após um

%        comando de atribuição; e

%      - imprimir uma condição em formato HTML.

%===============================================================================

%===============================================================================

%Descrição da consulta: declara os operadores para as condições contidas nas pré

%                       e pós-condições, e nos invariantes que não são 

%                       pré-definidos pelo SWI-Prolog.

%===============================================================================

:- op( 700, xfx, '<>' )  ,   % diferente

   op( 700, xfx, '<=' )  ,   % menor ou igual

   op( 699, xfy, '<=>' ) ,   % equivalência

   op( 698, xfy, '=>' )  ,   % implicação

   op( 697, xfy, 'or' )  ,   % operador booleano or (\/)

   op( 696, xfy, 'and' ) ,   % operador booleando and (/\) 

   op( 695, fy, 'not' )  .   % operador booleando not (~)

%===============================================================================

%Nome: lerCondicoes

%Descrição do predicado: lê as condições dadas pelo usuário para pré e pós-

%                        condições, e invariantes. 

%

%Variável de entrada: 

%   - Tokens: a lista de tokens ALP lida.

%

%Variáveis de saída:

%   - Invars:  a lista de invariantes para os loops do programa em ALP;

%   - PreCond: a pré-condição para o cálculo de hoare; e

%   - PosCond: a pós-condição para o cálculo de hoare.  

%===============================================================================

lerCondicoes( Tokens, Invars, PreCond, PosCond ) :-

   lerCondicao( PreCond, '\nPré-condição: ' )    ,  % lê pré-condição

   lerCondicao( PosCond, '\nPós-condição: ' )    ,  % lê pós-condição

   lerInvariantes( Tokens, Invars )              ,  % lê invariantes

   nl                                            ,

   !                                             .

%===============================================================================

%Nome: lerCondicao

%Descrição do predicado: lê uma condição dada pelo usuário para pré ou pós-

%                        condição, ou um invariante. 

%

%Variável de entrada: 

%   - Pedido: instrução a ser impressa na tela pedindo um tipo de condição ao

%             usuário.

%

%Variável de saída:

%   - Condicao: a condição dada pelo usuário. 

%===============================================================================

% pede um determinado tipo de condição ao usuário, a lê e checa sua validade. Se

% for inválida, pede novamente

lerCondicao( Condicao, Pedido )                       :-

   write( Pedido )                                    ,

   ( ( read( Condicao )                               ,

       checarCondicao( Condicao ) )                   ;

     ( write( '\nCondicao invalida! Tente de novo!' ) ,

       lerCondicao( Condicao, Pedido ) ) )            .

%===============================================================================

%Nome: lerInvariantes

%Descrição do predicado: lê os ivariantes para os loops encontrados no programa

%                        em ALP. 

%

%Variável de entrada: 

%   - Tokens: a lista de tokens ALP lida.

%

%Variável de saída:

%   - Invars: lista de invariantes, um para cada loop encontrado no programa. 

%===============================================================================

lerInvariantes( Tokens, Invars ) :-

   acharLoops( Tokens, Loops )   ,   % acha os loops do programa em ALP

   pedirInvars( Loops, Invars )  .   % lê os invariantes para os loops

%===============================================================================

%Nome: checarCondicao

%Descrição do predicado: determina se uma pré ou pós-condição, ou um invariante

%                        está com a sintaxe correta (expressão booleana).

%

%Variável de entrada: 

%   - Condicao: a pré ou pós-condição, ou o invariante que terá sua sintaxe 

%               checada.

%===============================================================================

% se a condição representada dentro do Prolog for uma variável livre (isso 

% ocorre se o usuário entra variáveis começadas com letras maiúsculas sem ' '),

% ela é considerada inválida

checarCondicao( Condicao ) :-

   var( Condicao )         ,

   !                       ,

   fail                    .

% testa se a condição é uma expressão booleana composta de conectivo binário:

% se-somente-se (<=>), se-então (=>), or ou and e checa recursivamente cada um

% dos operandos do conectivo recursivamente 

checarCondicao( Condicao )              :-

   ( Condicao = ( Cond_1 <=> Cond_2 )   ;

     Condicao = ( Cond_1 => Cond_2 )    ;

     Condicao = ( Cond_1 or Cond_2 )    ;

     Condicao = ( Cond_1 and Cond_2 ) ) ,

   checarCondicao( Cond_1 )             ,

   checarCondicao( Cond_2 )             .

% testa se a condição é uma condição negada (not Condicao) e checa 

% recursivamente a condição que vem depois do operador unário not (Condicao)

checarCondicao( not Condicao ) :-

   checarCondicao( Condicao )  . 

% testa se a condição é um átomo

checarCondicao( Condicao ) :-

   atom( Condicao )        .

% testa se a condição é uma expressão booleana obtida através de uma comparação

% entre duas expressões aritméticas: =, <>, <, >, <= ou >= e checa 

% recursivamente cada um dos operandos do conectivo recursivamente 

checarCondicao( Condicao )           :-

   ( Condicao = ( Exp_1 = Exp_2 )    ;

     Condicao = ( Exp_1 <> Exp_2 )   ;

     Condicao = ( Exp_1 < Exp_2 )    ;

     Condicao = ( Exp_1 > Exp_2 )    ;

     Condicao = ( Exp_1 <= Exp_2 )   ;

     Condicao = ( Exp_1 >= Exp_2 ) ) ,

   checarExpressao( Exp_1 )          ,

   checarExpressao( Exp_2 )          .

%===============================================================================

%Nome: checarExpressao

%Descrição do predicado: determina se uma expressão aritmética está com a 

%                        sintaxe correta (sintaxe igual a de um expressao 

%                        inteira em ALP).

%

%Variável de entrada: 

%   - Expressao: a expressão aritmética que terá sua sintaxe checada.

%===============================================================================

% se a expressão representada dentro do Prolog for uma variável livre (isso 

% ocorre se o usuário entra variáveis começadas com letras maiúsculas sem ' '),

% ela é considerada inválida

checarExpressao( Expressao ) :-

   var( Expressao )          ,

   !                         ,

   fail                      .

% testa se a expressão aritmética é composta de operador binário: +, -, * ou /

% e checa recursivamente cada um dos operandos da operação recursivamente 

checarExpressao( Expressao )         :-

   ( Expressao = ( Exp_1 + Exp_2 )   ;

     Expressao = ( Exp_1 - Exp_2 )   ;

     Expressao = ( Exp_1 * Exp_2 )   ;

     Expressao = ( Exp_1 / Exp_2 ) ) ,

   checarExpressao( Exp_1 )          ,

   checarExpressao( Exp_2 )          .       

% testa se a expressão aritmética é uma expressão negada (- Expressao) e checa 

% recursivamente a expressão que vem depois do sinal de menos (Expressao)

checarExpressao( - Expressao )  :-  

   checarExpressao( Expressao ) .

% testa se a expressão é um átomo (variável em ALP) ou um inteiro

checarExpressao( Expressao ) :-

   atom( Expressao )         ;

   integer( Expressao )      . 

%===============================================================================

%Nome: acharLoops

%Descrição do predicado: acha os loops existentes em uma lista de tokens lidos

%                        de um programa em ALP.

%

%Variável de entrada: 

%   - Tokens: a lista de tokens ALP lida.

%

%Variável de saída:

%   - Loops: os loops encontrados na lista de tokens ALP dada.

%

%Assertiva de entrada:

%   - A análise sintática do programa em ALP foi feita com sucesso.

%===============================================================================

% se todos os tokens já foram analisados, então já foram encontados todos os 

% loops (possivelmente nenhum), não havendo mais nenhum a ser encontrado

acharLoops( [ ], [ ] ) .

% se um token while for encontado, é porque foi encontrado o início de um loop,

% o qual será lido até o final, e o resto dos tokens serão analisados para se

% saber se há algum outro loop

acharLoops( [ while | Tokens ], [ while( Teste, Corpo ) | Loops ] )           :-

   serComando( while( Teste, Corpo ), [ while | Tokens ], [ while | Tokens ],

                                                          _Resto )            ,

   acharLoops( Tokens, Loops )                                                .

% se um token qualquer for encontado, a busca por loops continua no próximo 

% token

acharLoops( [ _Token | Tokens ], Loops ) :-

   acharLoops( Tokens, Loops )           .

%===============================================================================

%Nome: pedirInvars

%Descrição do predicado: pede ao usuário os invariantes para os loops do

%                        programa em ALP.

%

%Variável de entrada:

%   - Loops: a lista de loops do programa em ALP encontrados.

%

%Variável de saída:

%   - Invars: a lista de invariantes para os loops.

%===============================================================================

% se todos os loops já foram percorridos, então já foram lidos todos os 

% invariantes (possivelmente nenhum), não havendo mais nenhum invariante a ser

% lido

pedirInvars( [ ], [ ] ) .

% lê o invariante do primeiro loop e, recursivamente, lê os invariantes para os

% demais loops 

pedirInvars( [ Loop | Loops ], [ inv( Loop, Invar ) | Invars ] ) :-

   nl                                                            ,

   write( Loop )                                                 ,

   lerCondicao( Invar, '\nInvariante: ' )                        ,

   pedirInvars( Loops, Invars )                                  .

%===============================================================================

%Nome: procurarInvar

%Descrição do predicado: Procura o invariante para um dado loop.

%

%Variáveis de entrada:

%   - Invars:                a lista de pares < loop, invariante >; e

%   - while( Teste, Corpo ): o loop a ter o invariante procurado.

%

%Variável de saída:

%   - Invar: o invariante encontrado.

%===============================================================================

% se o par sendo analisado possui o loop cujo invariante é o que está sendo

% procurado, o invariante do par é a resposta

procurarInvar( [ inv( Loop, Invar ) | _Invars ], while( Teste, Corpo ),

                                                               Invar ) :-

   Loop == while( Teste, Corpo )                                       .

% caso contrário, o invariante é procurado nos demais pares

procurarInvar( [ _LoopInv | Invars ], while( Teste, Corpo ), Invar ) :-

   procurarInvar( Invars, while( Teste, Corpo ), Invar )             .

%===============================================================================

%Nome: serIgual

%Descrição do predicado: analisa se uma pré-condição é equivalente a uma pós-

%                        condição após um comando de atribuição.

%

%Variáveis de entrada:

%   - PreCond: a pré-condição;

%   - Var:     a variável que recebe um valor na atribuição Var := Exp;

%   - Exp:     a expressão que é atribuída na atribuição Var := Exp; e

%   - PosCond: a pós-condição.

%===============================================================================

% a pré-condição precisa ter a mesma forma de uma pós-condição para ser 

% equivalente a mesma após um comando de atribuição, e este predicado verifica 

% se ambas são da forma A op B, onde A e B também devem ser checados 

% recursivamente para ver se possuem esta propriedade

serIgual( PreCond, Var, Exp, PosCond )  :-

   ( ( PosCond = ( Exp_1 <=> Exp_2 )    ,   % forma A <=> B

       PreCond = ( Exp_3 <=> Exp_4 ) )  ;

     ( PosCond = ( Exp_1 => Exp_2 )     ,   % forma A => B

       PreCond = ( Exp_3 => Exp_4 ) )   ;

     ( PosCond = ( Exp_1 or Exp_2 )     ,   % forma A or B

       PreCond = ( Exp_3 or Exp_4 ) )   ;

     ( PosCond = ( Exp_1 and Exp_2 )    ,   % forma A and B

       PreCond = ( Exp_3 and Exp_4 ) )  ;

     ( PosCond = ( Exp_1 = Exp_2 )      ,   % forma A = B

       PreCond = ( Exp_3 = Exp_4 ) )    ;

     ( PosCond = ( Exp_1 <> Exp_2 )     ,   % forma A <> B

       PreCond = ( Exp_3 <> Exp_4 ) )   ;

     ( PosCond = ( Exp_1 < Exp_2 )      ,   % forma A < B

       PreCond = ( Exp_3 < Exp_4 ) )    ;

     ( PosCond = ( Exp_1 > Exp_2 )      ,   % forma A > B

       PreCond = ( Exp_3 > Exp_4 ) )    ;

     ( PosCond = ( Exp_1 <= Exp_2 )     ,   % forma A <= B

       PreCond = ( Exp_3 <= Exp_4 ) )   ;

     ( PosCond = ( Exp_1 >= Exp_2 )     ,   % forma A >= B

       PreCond = ( Exp_3 >= Exp_4 ) )   ;

     ( PosCond = ( Exp_1 + Exp_2 )      ,   % forma A + B

       PreCond = ( Exp_3 + Exp_4 ) )    ;

     ( PosCond = ( Exp_1 - Exp_2 )      ,   % forma A - B

       PreCond = ( Exp_3 - Exp_4 ) )    ;

     ( PosCond = ( Exp_1 * Exp_2 )      ,   % forma A * B

       PreCond = ( Exp_3 * Exp_4 ) )    ;

     ( PosCond = ( Exp_1 / Exp_2 )      ,   % forma A / B

       PreCond = ( Exp_3 / Exp_4 ) ) )  ,

   serIgual( Exp_3, Var, Exp, Exp_1 )   ,   % análise recursiva

   serIgual( Exp_4, Var, Exp, Exp_2 )   .   % análise recursiva

% a pré-condição precisa ter a mesma forma de uma pós-condição para ser 

% equivalente a mesma após um comando de atribuição, e este predicado verifica 

% se ambas são da forma not A, onde A também deve ser checado recursivamente 

% para ver se possui esta propriedade

serIgual( not PreCond, Var, Exp, not PosCond ) :-

   serIgual( PreCond, Var, Exp, PosCond )      . 

% a pré-condição precisa ter a mesma forma de uma pós-condição para ser

% equivalente a mesma após um comando de atribuição, e este predicado verifica 

% se ambas são da forma - A, onde A também deve ser checado recursivamente para 

% ver se possui esta propriedade

serIgual( - PreCond, Var, Exp, - PosCond ) :-

   serIgual( PreCond, Var, Exp, PosCond )  .

% neste caso, a pré-condição é equivalente a pós-condição após o comando de

% atribuição, com Exp sendo substituído por var na pré-condição

serIgual( Exp, Var, Exp, Var ) .

% neste caso, a pré-condição é equivalente a pós-condição após o comando de

% atribuição, porque são iguais e Var não aparece na pós-condição (se Var não

% aparece na pós-condição, nada tem que ser substituído)

serIgual( Atomo, Var, _Exp, Atomo ) :-

   Atomo \== Var                    ,

   ( atom( Atomo )                  ;

     integer( Atomo ) )             .

%===============================================================================

%Nome: imprimirCond

%Descrição do predicado: imprime uma pré ou pós-condição, invariante ou sentença

%                        a ser demonstrada.

%

%Variável de entrada:

%   Condição: a condição (pré ou pós-condição, invariante ou sentença) a ser

%             impressa.

%===============================================================================

% a condição a ser impressa é um átomo (variável ou as constantes true ou false)

% ou um inteiro

imprimirCond( Condicao )   :-

   ( atom( Condicao )      ;

     integer( Condicao ) ) ,

   write( Condicao )       .

% a condição a ser impressa é da forma A op B, onde A e B são impressos

% recursivamente, sendo a expressão impressa com a devida parentização

imprimirCond( Condicao )              :-

   ( ( Condicao = ( Exp_1 <=> Exp_2 ) ,   % forma A <=> B

       Op = ' <=> ' )                 ;

     ( Condicao = ( Exp_1 => Exp_2 )  ,   % forma A => B

       Op = ' => ' )                  ;

     ( Condicao = ( Exp_1 or Exp_2 )  ,   % forma A or B

       Op = ' or ' )                  ;

     ( Condicao = ( Exp_1 and Exp_2 ) ,   % forma A and B

       Op = ' and ' )                 ;

     ( Condicao = ( Exp_1 = Exp_2 )   ,   % forma A = B

       Op = ' = ' )                   ;

     ( Condicao = ( Exp_1 <> Exp_2 )  ,   % forma A <> B

       Op = ' <> ' )                  ;

     ( Condicao = ( Exp_1 < Exp_2 )   ,   % forma A < B

       Op = ' < ' )                   ;

     ( Condicao = ( Exp_1 > Exp_2 )   ,   % forma A > B

       Op = ' > ' )                   ;

     ( Condicao = ( Exp_1 <= Exp_2 )  ,   % forma A <= B

       Op = ' <= ' )                  ;

     ( Condicao = ( Exp_1 >= Exp_2 )  ,   % forma A >= B

       Op = ' >= ' )                  ;

     ( Condicao = ( Exp_1 + Exp_2 )   ,   % forma A + B

       Op = ' + ' )                   ;

     ( Condicao = ( Exp_1 - Exp_2 )   ,   % forma A - B

       Op = ' - ' )                   ;

     ( Condicao = ( Exp_1 * Exp_2 )   ,   % forma A * B

       Op = ' * ' )                   ;

     ( Condicao = ( Exp_1 / Exp_2 )   ,   % forma A / B

       Op = ' / ' ) )                 ,

   write( '( ' )                      ,

   imprimirCond( Exp_1 )              ,   % análise recursiva

   write( Op )                        ,

   imprimirCond( Exp_2 )              ,   % análise recursiva

   write( ' )' )                      .

% a condição a ser impressa é da forma - A, onde A é impresso recursivamente

imprimirCond( - Cond )  :-

   write( '- ' )        ,

   imprimirCond( Cond ) .

% a condição a ser impressa é da forma not A, onde A é impresso recursivamente

imprimirCond( not Cond ) :-

   write( 'not ' )       ,

   imprimirCond( Cond )  .

%===============================================================================

%NOME: expressoes.pl

%

%Projeto: Correção de Programas

%

%   Versão: V 1.00

%   Autora : Juliana Carpes Imperial - 0124811-7

%

%Descrição do módulo

%   Objetivo:

%      - converter uma expressão lida pelo analisador sintático para o formato

%        das condições utilizado durante a correção de programas.

%===============================================================================

%===============================================================================

%Nome: converter

%Descrição do predicado: converte uma expressão lida pelo analisador sintático

%                        para o formato das condições utilizado durante a

%                        correção de programas.

%

%Variável de entrada:

%   - a expressão a ser convertida.

%

%Variável de saída:

%   - a expressão convertida para o mesmo formato de uma condição.

%===============================================================================

converter( num( Num ), Num ) .  % converte um número

converter( ide( Var ), Var ) .  % converte uma variável

converter( true, true )      .  % converte a constante booleana true

converter( false, false )    .  % converte a constante booleana false

% converte uma soma, convertendo recursivamente seus operandos

converter( expInt( soma, Exp_1, Exp_2 ), Exp_3 + Exp_4 ) :-

   converter( Exp_1, Exp_3 )                             ,

   converter( Exp_2, Exp_4 )                             .

% converte uma diferença, convertendo recursivamente seus operandos

converter( expInt( menos, Exp_1, Exp_2 ), Exp_3 - Exp_4 ) :-

   converter( Exp_1, Exp_3 )                              ,

   converter( Exp_2, Exp_4 )                              .

% converte uma multiplicação, convertendo recursivamente seus operandos

converter( expInt( mult, Exp_1, Exp_2 ), Exp_3 * Exp_4 ) :-

   converter( Exp_1, Exp_3 )                             ,

   converter( Exp_2, Exp_4 )                             .

% converte uma divisão, convertendo recursivamente seus operandos

converter( expInt( divide, Exp_1, Exp_2 ), Exp_3 / Exp_4 ) :-

   converter( Exp_1, Exp_3 )                               ,

   converter( Exp_2, Exp_4 )                               .

% converte uma inversão de sinal, convertendo recursivamente a expressão a ter o

% sinal invertido

converter( menos( Exp_1 ), - Exp_2 ) :-

   converter( Exp_1, Exp_2 )         .

% converte um and lógico, convertendo recursivamente seus operandos

converter( expBool( and, Exp_1, Exp_2 ), Exp_3 and Exp_4 ) :-

   converter( Exp_1, Exp_3 )                               ,

   converter( Exp_2, Exp_4 )                               .

% converte um or lógico, convertendo recursivamente seus operandos

converter( expBool( or, Exp_1, Exp_2 ), Exp_3 or Exp_4 ) :-

   converter( Exp_1, Exp_3 )                             ,

   converter( Exp_2, Exp_4 )                             .

% converte uma negação, convertendo recursivamente a expressão negada

converter( neg( Exp_1 ), not Exp_2 ) :-

   converter( Exp_1, Exp_2 )         .

% converte uma comparação de igualdade, convertendo recursivamente seus 

% operandos

converter( expBool( igual, Exp_1, Exp_2 ), Exp_3 = Exp_4 ) :-

   converter( Exp_1, Exp_3 )                               ,

   converter( Exp_2, Exp_4 )                               .

% converte uma comparação de desigualdade, convertendo recursivamente seus 

% operandos

converter( expBool( difer, Exp_1, Exp_2 ), Exp_3 <> Exp_4 ) :-

   converter( Exp_1, Exp_3 )                                ,

   converter( Exp_2, Exp_4 )                                .

% converte uma comparação de maior, convertendo recursivamente seus operandos

converter( expBool( maior, Exp_1, Exp_2 ), Exp_3 > Exp_4 ) :-

   converter( Exp_1, Exp_3 )                               ,

   converter( Exp_2, Exp_4 )                               .

% converte uma comparação de menor, convertendo recursivamente seus operandos

converter( expBool( menor, Exp_1, Exp_2 ), Exp_3 < Exp_4 ) :-

   converter( Exp_1, Exp_3 )                                ,

   converter( Exp_2, Exp_4 )                                .

% converte uma comparação de maior ou igual, convertendo recursivamente seus 

% operandos

converter( expBool( maiorIg, Exp_1, Exp_2 ), Exp_3 >= Exp_4 ) :-

   converter( Exp_1, Exp_3 )                                  ,

   converter( Exp_2, Exp_4 )                                  .

% converte uma comparação de menor ou igual, convertendo recursivamente seus 

% operandos

converter( expBool( menorIg, Exp_1, Exp_2 ), Exp_3 <= Exp_4 ) :-

   converter( Exp_1, Exp_3 )                                  ,

   converter( Exp_2, Exp_4 )                                  .

%===============================================================================

%NOME: hoare.pl

%

%Projeto: Correção de Programas

%

%   Versão: V 1.00

%   Autora : Juliana Carpes Imperial - 0124811-7

%

%Descrição do módulo

%   Objetivos:

%      - implementação das regras de inferência do cálculo de Hoare para fazer

%        a correção de programas escritos em ALP;

%      - implementação de um filtro que retira provas de programas que possuem

%        o conjunto de sentenças a ser provado igual a de outras; e

%      - implementação de predicados para criar uma página HTML contendo várias

%        provas de programa utilizando o cálculo de Hoare juntamente com as

%        sentenças que precisam ser demonstradas para que as provas sejam

%        válidas.

%===============================================================================

%===============================================================================

%Nome: fazerHoare

%Descrição do predicado: aplica as regras de inferência do cálculo de hoare para

%                        fazer a correção do programa em ALP.

%

%Variáveis de entrada:

%   - Invars:   a lista de pares < loop, invariante >;

%   - PreCond:  a pré-condição para o trecho de programa sendo corrigido;

%   - Cmds:     o trecho de programa sendo corrigido;

%   - PosCond:  a pós-condição para o trecho de programa sendo corrigido; e

%   - SentsAnt: sentenças a serem demonstradas antes de aplicar a regra.

%

%Variáveis de saída:

%   - Prova:    a árvore de prova resultante da aplicação da regra; e

%   - SentsDep: sentenças a serem demonstradas depois da aplicação da regra.

%===============================================================================

% comando skip; regra é axioma: ________________

%                               { P } skip { P }

fazerHoare( _Invars, Cond, [ skip ], Cond, Prova, Sents, Sents ) :-

   Prova = [ [ Cond, [ skip ], Cond ] ]                          .

% comando de atribuição; regra é axioma: _________________________

%                                        { P[ x/E ] } x := E { P }

% pelo menos uma das condições deve ser totalmente ligada para que a regra

% funcione corretamente. A expressão no programa deve ser convertida para o

% formato de condição antes de tentar se chegar no axioma

fazerHoare( _Invars, PreCond, [ atribui( Var, Exp ) ], PosCond, Prova, Sents,

                                                                Sents ) :-

   ( checarCondicao( PreCond )                                          ;

     checarCondicao( PosCond ) )                                        ,

   converter( Exp, EquivExp )                                           ,

   serIgual( PreCond, Var, EquivExp, PosCond )                          ,

   Prova = [ [ PreCond, [ atribui( Var, Exp ) ], PosCond ] ]            .

% regra para mais de um comando: { P } CMD1 { R }  { R } CDM2 { Q }

%                                __________________________________

%                                     { P } CMD1 ; CMD2 { Q }

% deve haver mais de um comando para que a regra funcione corretamente. São

% tentadas duas alternativas, uma começando pelo primeiro comando e outra

% pelos demais comandos

fazerHoare( Invars, PreCond, [ Cmd | Cmds ], PosCond, Prova, SentsAnt,

                                                             SentsDep )       :-

   length( Cmds, TamCmds )                                                    ,

   TamCmds > 0                                                                ,

   ( ( fazerHoare( Invars, PreCond, [ Cmd ], CondTemp, ProvaFilha_1, SentsAnt ,

                                                       SentsTemp )            ,

       fazerHoare( Invars, CondTemp, Cmds, PosCond, ProvaFilha_2, SentsTemp,

                                                    SentsDep ) )              ;

     ( fazerHoare( Invars, CondTemp, Cmds, PosCond, ProvaFilha_2, SentsAnt,

                                                    SentsTemp )               ,

       fazerHoare( Invars, PreCond, [ Cmd ], CondTemp, ProvaFilha_1, SentsTemp,

                                                       SentsDep ) ) )         ,

   Prova = [ [ PreCond, [ Cmd | Cmds ], PosCond ], ProvaFilha_1,

                                                   ProvaFilha_2 ]             .

% regra do if sem else: { P and B } CMD { Q }  P and not B => Q

%                       _______________________________________

%                            { P } if B then CMD fi { Q }

% o teste do if deve ser convertido para o formato de condição antes da

% tentativa de aplicar a regra. A pós-condição não deve ser variável livre

% depois da chamada recursiva para que a sentença a ser demonstrada seja bem

% formada. Se a pré-condição for variável livre, tenta-se dois chutes com os

% dois possíveis casos do if. A sentença a ser demonstrada no if é incluída no

% conjunto de sentenças.

fazerHoare( Invars, PreCond, [ if( Teste, Entao ) ], PosCond, Prova, SentsAnt,

                                                              SentsDep )    :-

   converter( Teste, EquivTeste )                                           ,

   fazerHoare( Invars, PreCond and EquivTeste, Entao, PosCond, ProvaFilha,

                                                      SentsAnt, SentsTemp ) ,

   nonvar( PosCond )                                                        ,

   SentsDep = [ ( PreCond and not EquivTeste ) => PosCond | SentsTemp ]     ,

   Prova = [ [ PreCond, [ if( Teste, Entao ) ], PosCond ], ProvaFilha,

             [ ( PreCond and not EquivTeste ) => PosCond ] ]                ,

   ( var( PreCond )                                                         ->

     ( PreCond = EquivTeste                                                 ;

       PreCond = not EquivTeste                                             ;

       true )                                                               ;

     true )                                                                 .

% regra do if com else: { P and B } CMD1 { Q }  { P and not B } CMD2 { Q }

%                       __________________________________________________

%                         { P } if B then CMD1 else CMD2 esle_fi { Q }

% o teste do if deve ser convertido para o formato de condição antes da

% tentativa de aplicar a regra. Se a pré-condição for variável livre, tenta-se

% dois chutes com os dois possíveis casos do if.

fazerHoare( Invars, PreCond, [ if( Teste, Entao, Senao ) ], PosCond, Prova,

                                                 SentsAnt, SentsDep )       :-

   converter( Teste, EquivTeste )                                           ,

   fazerHoare( Invars, PreCond and EquivTeste, Entao, PosCond, ProvaFilha_1,

                                                      SentsAnt, SentsTemp ) ,

   fazerHoare( Invars, PreCond and not EquivTeste, Senao, PosCond,

                       ProvaFilha_2, SentsTemp, SentsDep )                  ,

   Prova = [ [ PreCond, [ if( Teste, Entao, Senao ) ], PosCond ],

             ProvaFilha_1, ProvaFilha_2 ]                                   ,

   ( var( PreCond )                                                         ->

     ( PreCond = EquivTeste                                                 ;

       PreCond = not EquivTeste                                             ;

       true )                                                               ;

     true )                                                                 .

% regra do while:          { P and B } CMD { P }

%                 _______________________________________

%                 { P } while B do CMD od { P and not B }

% o teste do while deve ser convertido para o formato de condição antes da

% tentativa de aplicar a regra.

fazerHoare( Invars, Invar, [ while( Teste, Corpo ) ], Invar and not EquivTeste,

                                           Prova, SentsAnt, SentsDep )     :-

   converter( Teste, EquivTeste )                                          ,

   fazerHoare( Invars, Invar and EquivTeste, Corpo, Invar, ProvaFilha,

                                                    SentsAnt, SentsDep )   ,

   Prova = [ [ Invar, [ while( Teste, Corpo ) ], Invar and not EquivTeste ],

             ProvaFilha ]                                                  .

% neste caso, a pré-condição é o invariante (P), mas a pós precisa ser

% fortalecida para ser igual a P and not B, para que a regra do while escrita

% acima possa ser aplicada. O teste do while deve ser convertido para o formato

% de condição antes da recursão e a pós-condição não pode ser variável livre

% para que a sentença a ser demonstrada seja bem formada. Esta é incluída no

% conjunto de sentenças.

fazerHoare( Invars, Invar, [ while( Teste, Corpo ) ], PosCond, Prova, SentsAnt,

                                                               SentsDep )  :-

   converter( Teste, EquivTeste )                                          ,

   PosCond \== ( Invar and not EquivTeste )                                ,

   nonvar( PosCond )                                                       ,

   fazerHoare( Invars, Invar, [ while( Teste, Corpo ) ],

               Invar and not EquivTeste, ProvaFilha, SentsAnt, SentsTemp ) ,

   SentsDep = [ ( Invar and not EquivTeste ) => PosCond | SentsTemp ]      ,

   Prova = [ [ Invar, [ while( Teste, Corpo ) ], PosCond ], ProvaFilha,

             [ ( Invar and not EquivTeste ) => PosCond ] ]                 .

% neste caso, a pré-condição não é o invariante (P), que antes precisa ser

% encontrado na lista de < loop, invariante > para que possa ser enfraquecido

% para ser igual a P, para que a regra do while escrita acima possa ser aplicada

% caso a pós também já esteja no formato correto. A pré-condição não pode ser

% variável livre para que a sentença a ser demonstrada seja bem formada. Esta é

% incluída no conjunto de sentenças.

fazerHoare( Invars, PreCond, [ while( Teste, Corpo ) ], PosCond, Prova,

                                                        SentsAnt, SentsDep ) :-

   procurarInvar( Invars, while( Teste, Corpo ), Invar )                     ,

   !                                                                         ,

   PreCond \== Invar                                                         ,

   ( var( PreCond )                                                          ->

     PreCond = Invar                                                         ;

     true )                                                                  ,

   fazerHoare( Invars, Invar, [ while( Teste, Corpo ) ], PosCond, ProvaFilha,

                                              SentsAnt, SentsTemp )          ,

   ( PreCond \== Invar                                                       ->

     ( SentsDep = [ PreCond => Invar | SentsTemp ]                           ,

       Prova = [ [ PreCond, [ while( Teste, Corpo ) ], PosCond ],

                 [ PreCond => Invar ], ProvaFilha ] )                        ;

     ( SentsDep = SentsTemp                                                  ,

       Prova = [ [ PreCond, [ while( Teste, Corpo ) ], PosCond ],

                 ProvaFilha ] ) )                                            .

% fortalecimento da pós-condição: { P } CMD { R }  R => Q

%                                 _______________________

%                                     { P } CMD { Q }

% as pré e pós-condições não podem ser variáveis livres, assim como a nova pós-

% condição após a chamada recursiva, que também deve ser diferente da antiga

% pós-condição. A sentença a ser demonstrada no fortalecimento é incluída no

% conjunto de sentenças.

fazerHoare( Invars, PreCond, [ Comando ], PosCond, Prova, SentsAnt,

                                                          SentsDep ) :-

   nonvar( PreCond )                                                 ,

   nonvar( PosCond )                                                 ,

   fazerHoare( Invars, PreCond, [ Comando ], NovaPosCond, ProvaFilha,

                                             SentsAnt, SentsTemp )   ,

   nonvar( NovaPosCond )                                             ,

   NovaPosCond \== PosCond                                           ,

   SentsDep = [ NovaPosCond => PosCond | SentsTemp ]                 ,

   Prova = [ [ PreCond, [ Comando ], PosCond ], ProvaFilha,

             [ NovaPosCond => PosCond ] ]                            .

% enfraquecimento da pré-condição: P => R  { R } CMD { Q }

%                                  _______________________

%                                      { P } CMD { Q }

% as pré e pós-condições não podem ser variáveis livres, assim como a nova pré-

% condição após a chamada recursiva, que também deve ser diferente da antiga

% pré-condição. A sentença a ser demonstrada no enfraquecimento é incluída no

% conjunto de sentenças.

fazerHoare( Invars, PreCond, [ Comando ], PosCond, Prova, SentsAnt,

                                                          SentsDep )       :-

   nonvar( PreCond )                                                       ,

   nonvar( PosCond )                                                       ,

   fazerHoare( Invars, NovaPreCond, [ Comando ], PosCond, ProvaFilha,

                                                 SentsAnt, SentsTemp )     ,

   nonvar( NovaPreCond )                                                   ,

   NovaPreCond \== PreCond                                                 ,

   SentsDep = [ PreCond => NovaPreCond | SentsTemp ]                       ,

   Prova = [ [ PreCond, [ Comando ], PosCond ], [ PreCond => NovaPreCond ] ,

             ProvaFilha ]                                                  .

%===============================================================================

%Nome: retirarProvas

%Descrição do predicado: retira provas da lista de respostas (uma lista de

%                        < provas, sentenças a serem demonstradas > de forma

%                        que todas possuam um conjunto de sentenças a serem

%                        demonstradas diferentes entre si.

%

%Variáveis de entrada:

%   - Respostas:  uma lista de < provas, sentenças a serem demonstradas >; e

%   - ListaSents: a lista conjunto de sentenças encontradas até então.

%

%Variável de saída:

%   - NovasResp: uma lista de < provas, sentenças a serem demonstradas > tal que

%                todas as provas possuem um conjunto de sentenças a serem

%                demonstradas diferentes entre si

%===============================================================================

% caso base, todas as provas já foram analisadas com relação às suas sentenças

retirarProvas( [ ], [ ], _ListaSents ) .

% o conjunto de sentenças da prova não é igual ao de nenhuma das provas já

% analisadas. A chamada recursiva coloca o conjunto desta para as próximas

% análises.

retirarProvas( [ [ Prova, Sents ] | Respostas ],

               [ [ Prova, Sents ] | NovasResp ], ListaSents )    :-

   \+ member( Sents, ListaSents )                                ,

   retirarProvas( Respostas, NovasResp, [ Sents | ListaSents ] ) .

% o conjunto de sentenças da prova é igual ao de uma das provas já analisadas.

% Esta prova não fará parte do conjunto de provas resultante.

retirarProvas( [ _Resposta | Respostas ], NovasResp, ListaSents ) :-

   retirarProvas( Respostas, NovasResp, ListaSents )              .

%===============================================================================

%Nome: imprimirSaida

%Descrição do predicado: cria uma página HTML contendo várias provas de programa

%                        utilizando o cálculo de Hoare juntamente com as

%                        sentenças que precisam ser demonstradas para que as

%                        provas sejam válidas.

%

%Variável de entrada:

%   - Respostas: a lista de < provas, sentenças a serem demonstradas > com todas

%                as provas possuindo um conjunto de sentenças a serem

%                demonstradas diferentes entre si.

%

%Assertiva de saída:

%   - a página HTML foi criada com sucesso e seu arquivo foi fechado, ou foi

%     detectado um erro de abertura, escrita ou fechamento de arquivo o que leva

%     o programa a ser abortado.

%===============================================================================

% tenta abrir o arquivo da página HTML para escrita. Se der erro, este é

% tratado. Se o arquivo for aberto com sucesso, seu cabeçalho é escrito,

% juntamente com as provas e seus conjuntos de sentenças devidamente formatados,

% e, por fim, são escritas as tags do fim da página. Se todo este processo de

% escrita no arquivo for feito com sucesso, ele é fechado. Se der algum erro de

% escrita ou fechamento, este é tratado

imprimirSaida( Respostas )                                                    :-

   catch( tell( 'saida.html' ), _Excecao, tratarErro( abertura ) )            ,

   catch( write( '<HTML>\n<HEAD> ' ), _Excecao, tratarErro( escrita ) )       ,

   catch( write( '<TITLE> Prova de ' ), _Excecao, tratarErro( escrita ) )     ,

   catch( write( 'Correção de Programas' ), _Excecao, tratarErro( escrita ) ) ,

   catch( write( ' </TITLE> </HEAD>\n' ), _Excecao, tratarErro( escrita ) )   ,

   catch( write( '<BODY> <B> <PRE>\n' ), _Excecao, tratarErro( escrita ) )    ,

   catch( imprimirProvas( Respostas ), _Excecao, tratarErro( escrita ) )      ,

   catch( write( '\n</B> </PRE> </BODY>' ), _Excecao, tratarErro( escrita ) ) ,

   catch( write( '\n<HTML>' ), _Excecao, tratarErro( escrita ) )              ,

   catch( told, _Excecao, tratarErro( fechamento ) )                          .

%===============================================================================

%Nome: imprimirProvas

%Descrição do predicado: imprime as provas e suas sentenças a serem demonstradas

%                        com formatação HTML.

%

%Variável de entrada:

%   - Respostas: a lista de < provas, sentenças a serem demonstradas > com todas

%                as provas possuindo um conjunto de sentenças a serem

%                demonstradas diferentes entre si.

%

%Assertiva de entrada:

%   - O arquivo da página HTML está aberto.

%===============================================================================

% todas as provas e seus conjuntos de sentenças já foram impressos

imprimirProvas( [ ] ) .

% imprime uma prova em cálculo de hoare e o conjunto de sentenças a ser

% demonstrado dela caso não seja vazio

imprimirProvas( [ [ Prova, Sents ] | Respostas ] )        :-

   write( '<H4>Prova: </H4>\n' )                          ,

   imprimirHoare( Prova, 1 )                              ,

   ( Sents \== [ ]                                        ->

     ( write( '\n<H4>Sentenças a serem provadas: </H4>' ) ,

       imprimirSents( Sents ) )                           ;

     true )                                               ,

   ( Respostas \== [ ]                                    ->

     write( '\n\n<HR>' )                                  ;

     true )                                               ,

   imprimirProvas( Respostas )                            .

%===============================================================================

%Nome: imprimirHoare

%Descrição do predicado: imprime as provas de correção de programa com

%                        formatação HTML.

%

%Variáveis de entrada:

%   - Provas: a árvore com a prova de correção de programa; e

%   - Nivel:  o nível que está sendo impresso a partir da raiz da árvore.

%

%Assertiva de entrada:

%   - O arquivo da página HTML está aberto.

%===============================================================================

% está se imprimindo um axioma, que é um trecho de prova sem filhos (uma folha

% da árvore), o que inclui as regras do skip e de atribuição

imprimirHoare( [ ProvaPai ], Nivel ) :-

   imprimirProva( ProvaPai, Nivel )  .

% está se imprimindo um ramo de prova que só tem um filho (while). O nível de um

% filho é sempre maior do que o do pai em um

imprimirHoare( [ ProvaPai, ProvaFilha ], Nivel ) :-

   imprimirProva( ProvaPai, Nivel )              ,

   imprimirHoare( ProvaFilha, Nivel + 1 )        .

% está se imprimindo um ramo de prova que só tem dois filhos (regras do if,

% enfraquecimento de pré-condição, fortalecimento de pós-condição e para mais de

% um comando). O nível de um filho é sempre maior do que o do pai em um

imprimirHoare( [ ProvaPai, ProvaFilha_1, ProvaFilha_2 ], Nivel ) :-

   imprimirProva( ProvaPai, Nivel )                              ,

   imprimirHoare( ProvaFilha_1, Nivel + 1 )                      ,

   imprimirHoare( ProvaFilha_2, Nivel + 1 )                      .

%===============================================================================

%Nome: imprimirProva

%Descrição do predicado: imprime um ramo de prova de correção de programa com

%                        formatação HTML.

%

%Variáveis de entrada:

%   - Ramo:  um ramo da árvore de uma prova de correção de programa; e

%   - Nivel: o nível que está sendo impresso a partir da raiz da árvore.

%

%Assertiva de entrada:

%   - O arquivo da página HTML está aberto.

%===============================================================================

% o ramo é um trecho se programa sendo corrigido da forma { P } CMD { Q }

imprimirProva( [ PreCond, Prog, PosCond ], Nivel ) :-

   tab( ( Nivel - 1 ) * 3 )                        ,

   write( '<FONT COLOR = "red">{ ' )               ,

   imprimirCond( PreCond )                         ,

   write( ' } </FONT>' )                           ,

   write( Prog )                                   ,

   write( '<FONT COLOR = "red"> { ' )              ,

   imprimirCond( PosCond )                         ,

   write( ' }</FONT>\n' )                          .

% o ramo é uma sentença a ser demonstrada da forma P => Q

imprimirProva( Condicao, Nivel )     :-

   Condicao = ( _Cond_1 => _Cond_2 ) ,

   tab( ( Nivel - 1 ) * 3 )          ,

   write( '<FONT COLOR = "blue">' )  ,

   imprimirCond( Condicao )          ,

   write( ' </FONT>' )               ,

   nl                                .

%===============================================================================

%Nome: imprimirSents

%Descrição do predicado: imprime uma lista de sentenças, uma em cada linha.

%

%Variável de entrada:

%   - Sents: a lista de sentenças a ter seus elementos impressos.

%===============================================================================

imprimirSents( [ ] ) .                 % a lista já foi totalmente impressa

imprimirSents( [ Sent | Sents ] ) :-

   nl                             ,    % pula linha

   imprimirCond( Sent )           ,    % imprime uma sentença

   imprimirSents( Sents )         .    % imprime o resto da lista recursivamente

%===============================================================================

%NOME: erro.pl

%

%Projeto: Correção de Programas

%

%   Versão: V 1.00

%   Autora : Juliana Carpes Imperial - 0124811-7

%

%Descrição do módulo

%   Objetivo:

%      - tratar erro durante as operações de abertura, leitura, escrita e

%        fechamento em arquivos.

%===============================================================================

%===============================================================================

%Nome: tratarErro

%Descrição do predicado: trata o erro que aconteceu durante as operações com 

%                        arquivos.

%

%Variável de entrada: 

%   - Erro: o tipo do erro que aconteceu.

%

%Assertiva de entrada:

%   - o arquivo com o programa ALP está aberto.

%

%Assertiva de saída:

%   - o arquivo com o programa ALP está fechado e o programa foi abortado.

%===============================================================================

% exibe as mensagens adequadas para cada tipo de erro (abertura, leitura,

% escrita e fechamento de arquivo) e, por fim, fecha o arquivo e aborta o

% programa

tratarErro( Erro )                                              :-

   catch( seen, _Excecao, true )                                ,

   catch( told, _Excecao, true )                                , 

   ( ( Erro == abertura                                         , 

       write( '\nErro durante a abertura do arquivo.\n' ) )     ;

     ( Erro == escrita                                          ,

       write( '\nErro durante a escrita no arquivo.\n' ) )      ;

     ( Erro == leitura                                          ,

       write( '\nErro durante a leitura do arquivo.\n' ) )      ;

     ( Erro == fechamento                                       ,

       write( '\nErro durante o fechamento do arquivo.\n' ) ) ) ,

   abort                                                        .

%===============================================================================

%NOME: principal.pl

%

%Projeto: Correção de Programas

%

%   Versão: V 1.00

%   Autora : Juliana Carpes Imperial - 0124811-7

%

%Descrição do módulo

%   Objetivo:

%      - ser o módulo principal do corretor de programas.

%===============================================================================

%===============================================================================

%Descrição da consulta: carrega os demais arquivos com o código-fonte do 

%                       corretor de programas no interpretador de Prolog.

%===============================================================================

:- [ 'erro.pl', 'lexica.pl', 'sintatica.pl', 'condicoes.pl', 'expressoes.pl' ] , 

   [ 'hoare.pl' ]                                                              .

%===============================================================================

%Nome: fazerAnalise

%Descrição do predicado: executa todos os passos necessários para realizar a

%                        correção formal de um programa.

%===============================================================================

% faz a análise léxica e sintática de um programa. Posteriormente, lê as pré e

% pós condições, e os invariantes. Depois, aplica o cálculo de hoare no programa

% para provar a sua correção. Por fim, pega as várias provas diferentes obtidas

% e retira algumas de forma que as resultantes tenham o conjunto de sentenças a

% ser provado diferentes entre si, colocando-as em uma página HTML

corrigirPrograma                                                             :-

   fazerAnaLexica( Tokens )                                                  ,

   fazerAnaSintatica( Tokens, Arvore )                                       ,

   lerCondicoes( Tokens, Invars, PreCond, PosCond )                          ,

   setof( [ Prova, Sents ],

            fazerHoare( Invars, PreCond, Arvore, PosCond, Prova, [ ], Sents ),

            Respostas )                                                      ,

   retirarProvas( Respostas, NovasResp, [ ] )                                ,

   imprimirSaida( NovasResp )                                                .

4) ROTEITO DE TESTES EFETUADOS:

4.1) Critérios de Teste Efetuado:


Os critérios de testes adotados foram os seguintes:

1- Testes de abertura do arquivo de definição (o que acontece se é entrado um nome de um arquivo que não exista);

2- Testes que mostram que o programa só aceita arquivos se somente se tiverem o  trecho de código que esteja léxica e sintaticamente corretos. Foram feitos vários testes mostrando diversos erros desta categoria que podem ocorrer;

3- Testes com a leitura de condições (pré e pós-condições e invariantes); e

4- Testes com vários trechos de código corretos de acordo com a linguagem utilizada pelo programa e com diferentes pré e pós-condições, e invariantes, para testar as provas produzidas pelo programa e a geração do HTML. Estas devem estar semanticamente corretas de acordo com o cálculo de Hoare e satisfazendo os requisitos 7 e 8.

Os testes foram feitos utilizando a versão 4.0.9 do SWI-Prolog para Windows da University of Amsterdam, em um Pentium III 500 MHz, com 256 Mb de RAM com sistema operacional Windows 98, Second Edition. 


Infelizmente, não é possível testar a ocorrência de erros de leitura e fechamento de arquivo para ver como o programa se comporta, pois são condições excepcionais.
4.2) Descrição dos Casos Teste:


Antes de começar a descrição dos casos teste, é importante dizer que determinados erros no formato de dados entrados pelo usuário no prompt do SWI-Prolog são tratados automaticamente pelo interpretador. Como exemplo, incluem-se, mas não se limitam a, nomes de arquivos sem as aspas simples antes e depois do nome caso ele possua uma extensão, condições mal-formadas sem parentização para evitar certas ambigüidades, com operadores inexistentes ou aspas não fechadas ou incorretamente fechadas. Nestes casos, o interpretador aborta a execução do programa e emite uma mensagem de erro.

1- Se for entrado um nome de arquivo que não exista no diretório onde está localizado os arquivos .pl do programa ou no caminho especificado, o programa será abortado. Sendo que, antes disso aparecerá a seguinte mensagem de erro: 

Erro durante a abertura do arquivo. 


Caso não haja erro na abertura do arquivo, sempre o programa vai começar a análise léxica do trecho de código, como esperado.

2- Foram gerados arquivos com erros léxicos de dois tipos, que são os caracteres que não estão no conjunto dos que são aceitos pela linguagem (ex: @), e variáveis começando ou só com o caractere _ (a linguagem na qual o trecho de código deve estar escrita não permite isso). Nesses casos, o programa foi abortado e antes disso aparecia a mensagem:

Caractere ilegal: C, sendo C o caractere ilegal.

Em todos os outros casos testados, a análise léxica do trecho de código foi feita com sucesso, como deveria ser, seguindo o programa para a análise sintática.


Já para análise sintática, foram gerados erros sintáticos de vários tipos (de acordo com a definição da linguagem usada para o trecho de código que está no manual do usuário), tais como ponto e vírgula depois do último comando, ausência de ponto e vírgula entre dois comandos, comandos mal-formados, expressões mal-formadas, o token : sem ser seguido pelo = (:= é o sinal de atribuição), etc. Foram gerados diveros casos de erros deste tipo para saber se realmente o programa aceitava somente o que estava sintaticamente correto e qual o comportamento quando detectava um erro. Toda vez que havia um erro sintático e somente quando um acontecia, o programa foi abortado e exibiu a seguinte mensagem de erro:

Tokens reconhecidos: L1

Houve erro aqui: L2

onde L1 e L2 são duas listas de tokens. Em L1 estão os tokens que foram reconhecidos pelo analisador sintático e na segunda lista os que não foram reconhecidos. O erro sintático apareceu sempre na segunda lista, logo no início da mesma (nos primeiros elementos), como esperado.

3- Apenas as condições (pré e pós-condições e invariantes) bem formadas devem ser aceitas. Mal-formações sintáticas como as descritas logo no início desta seção já são tratadas pelo próprio interpretador. Tirando esses casos, somente condições com o formato de condição (expressão booleana) devem ser aceitas. Por exemplo, foram rejeitadas pelo programa os seguintes tipos de condições:

· aquelas que continham uma variável começada com letra maiúscula sem aspas simples antes e depois da mesma (isso é necessário para que o Prolog não a interprete como variável livre); e

· expressões inteiras ao invés de booleanas (expressões que não possuem valor verdade), como por exemplo a + b, a - b, a * b, a / b, - a, um número e os demais tipos de expressões inteiras. Se parte da expressão booleana contiver uma expressão inteira que não esteja em uma comparação (com os operadores =, <>, >, <, >= e <=) ela também não foi aceita, o que é bastante razoável. Ainda, expressões booleanas não foram aceitas em comparações, ou seja, utilizando os símbolos de <> e =, tal como na especificação do formato de condições.

Em ambos os casos, a condição foi pedida novamente ao usuário, aparecendo antes a seguinte mensagem:

Condicao invalida! Tente de novo!  


O programa só pára de pedir uma determinada condição quando uma no formato correto é dada. Uma condição com o formato é aquela que está no formato dado no manual do usuário. Todas as condições com este formato foram devidamente aceitas. Com todas as condições lidas e aceita, o programa começou a aplicar as regras do cálculo de Hoare. Exemplos de condições aceitas estão no próximo grupo de testes.

4- Foram feitos vários testes com diferentes programas léxica e sintaticamente corretos com diversas pré e pós-condições e invariantes. Na tabela abaixo, estão todos os testes feitos. Como as provas resultantes podem ficar muito grandes na horizontal, elas não foram incluídas neste documento, estando todas no disquete junto com os fontes. Em todos os casos, as provas obtidas estavam semanticamente corretas de acordo com o cálculo de Hoare e os requisitos 7 e 8 foram sempre respeitados, como deveria ser. Nos testes foi dado ênfase aos casos que caem no requisito 7 por serem os de maior interesse. Ainda, em todos os casos, em cada saída, todas as provas possuiam conjuntos de sentenças a serem demonstradas diferentes entre si, como desejado.

Na última coluna, está indicado se a pós-condição é derivada da pré depois da aplicação do trecho de código ou não (para saber se de fato os requisitos citados anteriormente são de fato satisfeitos, o que sempre ocorreu).


Os testes estão organizados de tal forma que os trechos de código a serem analisados estão ordenados do mais simples para o mais complexo.



	trecho de código
	pré-condição
	pós-condição
	arquivo de saída
	derivável

	teste01.txt
	true
	true
	saida01a.html
	sim

	
	0 = 0
	true
	saida01b.html
	sim

	
	true
	0 = 0
	saida01c.html
	sim

	teste02.txt
	true
	true
	saida02a.html
	sim

	
	0 = 0
	true
	saida02b.html
	sim

	
	true
	0 = 0
	saida02c.html
	sim

	
	true
	‘A’ = 5
	saida02d.html
	sim

	
	5 = 5
	‘A’ = 5
	saida02e.html
	sim

	
	5 = 5
	5 = 5
	saida02f.html
	sim

	
	‘A’ = ‘A’
	‘A’ = ‘A’
	saida02g.html
	sim

	
	5 = 4
	‘A’ = 4
	saida02h.html
	sim

	
	true
	( 5 = ‘A’ ) or ( ‘A’ = 4 )
	saida02i.html
	sim

	
	true
	( ‘A‘ = 5 ) and not ( ‘A’ <> 5 )
	saida02j.html
	sim

	
	true
	( ‘A’ = 5 ) and ( ‘A’ <> 4 )
	saida02k.html
	sim

	teste03.txt
	true
	a = b + c + d
	saida03a.html
	sim

	
	true
	a = c + b + d
	saida03b.html
	sim

	
	b + c + d 

= c + b + d
	a = c + d + b
	saida03c.html
	sim

	teste04.txt
	true
	true
	saida04a.html
	sim

	
	true
	a
	saida04b.html
	sim

	
	true
	not a
	saida04c.html
	não

	
	true
	a or not a
	saida04d.html
	sim

	
	not a
	true
	saida04e.html
	sim

	teste05.txt
	true
	true
	saida05a.html
	sim

	
	0 = 0
	true
	saida05b.html
	sim

	
	true
	0 = 0
	saida05c.html
	sim

	
	0 = 1
	1 = 2
	saida05d.html
	sim

	teste06.txt
	true
	true
	saida06a.html
	sim

	
	0 = 0
	true
	saida06b.html
	sim

	
	true
	0 = 0
	saida06c.html
	sim

	
	true
	a = 5
	saida06d.html
	sim

	
	5 = 5
	a = 5
	saida06e.html
	sim


	trecho de código
	pré-condição
	pós-condição
	arquivo de saída
	derivável

	teste07.txt
	true
	true
	saida07a.html
	sim

	
	0 = 0
	true
	saida07b.html
	sim

	
	true
	0 = 0
	saida07c.html
	sim

	
	true
	a = 5
	saida07d.html
	sim

	
	5 = 5
	a = 5
	saida07e.html
	sim

	teste08.txt
	true
	true
	saida08a.html
	sim

	
	true
	a = 5
	saida08b.html
	sim

	
	true
	b = 6
	saida08c.html
	sim

	
	true
	( a = 5 ) and ( b = 6 )
	saida08d.html
	sim

	
	true
	a = 6
	saida08e.html
	não

	
	true
	b = 5
	saida08f.html
	não

	
	true
	a + b = 11
	saida08g.html
	sim

	teste09.txt
	true
	true
	saida09a.html
	sim

	
	true
	a = 5
	saida09b.html
	sim

	
	true
	b = 6
	saida09c.html
	sim

	
	true
	( a = 5 ) and ( b = 6 )
	saida09d.html
	sim

	
	true
	a = 6
	saida09e.html
	não

	
	true
	b = 5
	saida09f.html
	não

	
	true
	11 = a + b
	saida09g.html
	sim

	teste10.txt
	true
	true
	saida10a.html
	sim

	
	true
	a = 5
	saida10b.html
	sim

	
	true
	b = 6
	saida10c.html
	sim

	
	true
	( a = 5 ) and ( b = 6 )
	saida10d.html
	sim

	
	true
	a = 6
	saida10e.html
	não

	
	true
	b = 5
	saida10f.html
	não

	
	true
	b + a = 11
	saida10g.html
	sim

	teste11.txt
	true
	true
	saida11a.html
	sim

	
	true
	a = 5
	saida11b.html
	sim

	
	true
	b = 6
	saida11c.html
	sim

	
	true
	( a = 5 ) and ( b = 6 )
	saida11d.html
	sim

	
	true
	a = 6
	saida11e.html
	não

	
	true
	b = 5
	saida11f.html
	não

	
	true
	11 = b + a
	saida11g.html
	sim

	teste12.txt
	true
	true
	saida12a.html
	sim

	
	true
	a = 5
	saida12b.html
	sim

	
	true
	b = 6
	saida12c.html
	sim

	
	true
	c = 11
	saida12d.html
	sim

	
	true
	( a = 5 ) and ( b = 6 ) and ( c = 11 )
	saida12e.html
	sim

	
	true
	a + b + c = 22
	saida12f.html
	sim

	
	true
	b + c + a = 22
	saida12g.html
	sim


obs: Como os trechos de códigos contidos nos arquivos teste08.txt ao teste11.txt produzem o mesmo resultado, é bastante razoavel que as mesmas pós-condições fossem derivadas das mesmas pré-condições após a execução dos códigos.

	trecho de código
	pré-condição
	pós-condição
	arquivo de saída
	derivável

	teste13.txt
	true
	true
	saida13a.html
	sim

	
	x = 0
	x = 1
	saida13b.html
	sim

	
	x = 0
	x <> 1
	saida13c.html
	não

	
	x <> 0
	x = 1
	saida13d.html
	não

	
	x <> 0
	x <> 1
	saida13e.html
	não

	
	x = 1
	x = 1
	saida13f.html
	sim

	
	x = 1
	x <> 1
	saida13g.html
	não

	
	true
	( x = 1 ) or ( x <> 1 )
	saida13h.html
	sim

	teste14.txt
	true
	true
	saida14a.html
	sim

	
	x = 0
	x = 1
	saida14b.html
	sim

	
	x = 0
	x <> 1
	saida14c.html
	não

	
	x <> 0
	x = 1
	saida14d.html
	não

	
	x <> 0
	x <> 1
	saida14e.html
	não

	
	x = 1
	x = 1
	saida14f.html
	sim

	
	x = 1
	x <> 1
	saida14g.html
	não

	
	true
	( x = 1 ) or ( x <> 1 )
	saida14h.html
	sim

	teste15.txt
	true
	true
	saida15a.html
	sim

	
	x > 0
	x = 1
	saida15b.html
	sim

	
	x <= 0
	x = -1
	saida15c.html
	sim

	
	true
	( x = 1 ) or ( x = -1 )
	saida15d.html
	sim

	teste16.txt
	true
	true
	saida16a.html
	sim

	
	a = 0
	( a = 1 ) or ( a = -1 )
	saida16b.html
	sim

	
	a > 0
	a >= 0
	saida16c.html
	sim

	
	a < 0
	a <= 0
	saida16d.html
	sim

	
	a = 0
	( ( b > 0 ) => ( a = -1 ) ) 

and ( ( b <= 0 ) => ( a = 1 ) )
	saida16e.html
	sim

	
	a > 0
	( ( b > 0 ) => ( a >= 0 ) ) 

and ( ( b <= 0 ) => ( a > 1 ) )
	saida16f.html
	sim

	
	a < 0
	( ( b > 0 ) => ( a < -1 ) ) 

and ( ( b <= 0 ) => ( a <= 0 ) )
	saida16g.html
	sim

	
	a >= 0
	( ( b > 0 ) => ( a >= -1 ) ) 

and ( ( b <= 0 ) => ( a > 0 ) )
	saida16h.html
	sim

	
	a <= 0
	( ( b > 0 ) => ( a < 0 ) ) 

and ( ( b <= 0 ) => ( a <= 1 ) )
	saida16i.html
	sim

	teste17.txt
	true
	true
	saida17a.html
	sim

	
	true
	x = 1
	saida17b.html
	sim

	
	true
	x = -1
	saida17c.html
	não

	
	true
	( x = 1 ) or ( x = -1 )
	saida17d.html
	sim

	
	x = 3
	x = 1
	saida17e.html
	sim

	teste18.txt
	true
	true
	saida18a.html
	sim

	
	true
	x = 2
	saida18b.html
	não

	
	true
	x = -2
	saida18c.html
	não

	
	true
	( x = 2 ) or ( x = -2 )
	saida18d.html
	sim

	
	x = 3
	x = 2
	saida18e.html
	sim


obs: Como os trechos de códigos contidos nos arquivos teste13.txt e teste14.txt produzem o mesmo resultado, é bastante razoavel que as mesmas pós-condições fossem derivadas das mesmas pré-condições após a execução dos códigos.

	trecho de código
	pré-condição
	pós-condição
	arquivo de saída
	derivável


	teste19.txt
	true
	true
	saida19a.html
	sim

	
	a and b or c
	a and b
	saida19b.html
	não

	
	a and b or c
	a and not b
	saida19c.html
	sim

	
	a and b
	a and not b
	saida19d.html
	sim

	
	c
	a and not b
	saida19e.html
	sim


	trecho de código
	pré-condição
	pós-condição
	invariante
	arquivo de saída
	derivável

	teste20.txt
	true
	true
	true
	saida20a.html
	sim

	
	true
	true
	a
	saida20b.html
	sim

	
	a
	a
	true
	saida20c.html
	sim

	
	a
	b
	true
	saida20d.html
	sim

	teste21.txt
	true
	true
	true
	saida21a.html
	sim

	
	true
	y >= x
	true
	saida21b.html
	sim

	
	y < x
	y = x
	y <= x
	saida21c.html
	sim

	
	y <= x
	y = x
	y <= x
	saida21d.html
	sim

	
	y = x
	x = y
	y <= x
	saida21e.html
	sim

	teste22.txt
	true
	true
	true
	saida22a.html
	sim

	
	true
	x = 10
	x <= 10
	saida22b.html
	sim

	
	true
	x = 10
	true
	saida22c.html
	não


	trecho de código
	pré-condição
	pós-condição
	invariante 1
	invariante 2
	arquivo de saída
	derivável

	teste23.txt
	true
	true
	true
	true
	saida23a.html
	sim

	
	a
	a
	true
	true
	saida23b.html
	sim

	
	a
	b
	true
	true
	saida23c.html
	sim

	teste24.txt
	y = 0
	y = 0
	y <= 10
	y >= 0
	saida24a.html
	sim


5) DOCUMENTAÇÃO PARA O USUÁRIO:
1- Em primeiro lugar, é necessário ter um interpretador de Prolog, o qual deve ser o SWI-Prolog ou um outro 100 % compatível com esse. O SWI-Prolog pode ser baixado em http://www.swi-prolog.org/. Todo este manual será baseado no uso do SWI-Prolog para Windows. Para utilizar um outro interpretador Prolog ou o SWI para outra plataforma, deve-se consultar o manual do mesmo para ver detalhes específicos de como usar o interpretador para rodar o programa.
2- Estando o interpretador corretamente instalado, deve-se criar um arquivo com o trecho de código a ser corrigido. Este deve estar em um arquivo somente texto, o qual pode ser criado utilizando o Notepad do Windows, por exemplo.
3- O trecho de código a ser corrigido deve estar escrito na linguagem ALP, cuja BNF é dada a seguir. Trechos de código que não respeitem esta sintaxe não serão aceitos pelo programa. Um trecho de código é uma lista de comandos (cmds). 
< cmds > ::= < cmd > | < cmd > ; < cmds >

< cmd > ::= < var > := < exp > | skip | if < bexp > then < cmds > fi
          | if < bexp > then < cmds > else < cmds > esle_fi

          | while < bexp > do < cmds > od

< exp > ::= < bexp > | < iexp >

< iexp > ::= < iterm > | < iterm > < fraco > < iexp >

< iterm > ::= < ielem > | < ielem > < forte > < iterm >

< ielem > ::= < num > | < var > | - ( < iexp > ) | ( < iexp > )
< bexp > ::= < bterm > | < bterm > or < bexp >

< bterm > ::= < belem > | < belem > and < bterm >

< belem > ::= true | false | < var > | not ( < bexp > ) 

            | ( < bexp > ) | < comp >

< comp > ::= < iexp > < rel > < iexp > 

< fraco > ::= + | -
< forte > ::= * | /
< rel > ::= <= | < | = | >= | <>
< var > ::= < letra > | < var > < letra > | < var > < digito >

          | < var > _
< num > ::= < digito > | < digito > < num >

< letra > ::= A | ... | Z | a | ... | z

< digito > ::= 0 | ... | 9
4- Para carregar o programa no SWI-Prolog para Windows, basta estar com a pasta onde os fontes se encontram aberta e dar dois cliques no arquivo principal.pl. Ou então, se o SWI-Prolog já estiver aberto, deve-se digitar o sequinte no prompt do interpretador:
[ ‘caminho_do_programa/principal.pl’ ] .

As aspas simples são necessárias se o código fonte do programa não estiver no diretório do Prolog e o caminho for necessário (senão, basta escrever apenas o nome do arquivo). O ponto no final deve ser sempre inserido quando se digitar algo no prompt do interpretados, como neste e em todos os outros casos a seguir. 

5- Para o programa funcionar corretamente, os seguintes arquivos devem estar presentes, todos juntos, em um único diretório:
· lexica.pl
· sintatica.pl
· condicoes.pl
· expressoes.pl
· hoare.pl
· erro.pl
· principal.pl
6- Com o programa carregado no interpretador, digite, no prompt do mesmo, o predicado principal do programa:
corrigirPrograma .
7- O programa vai pedir o arquivo que contém o código fonte a ser corrigido. Este deve ser entrado entre aspas simples caso possua extensão, e estar no diretório atual, desta maneira:
nome_arq . ou ‘nome_arq.ext’ .
8- Se o programa for abortado por causa de algum erro léxico ou sintático, ele deve ser corrigido no arquivo que contém o trecho de código para tentar de novo, a partir do passo 6. Se o erro for léxico, o programa apontará o caractere inválido. Já se for sintático, o programa mostrará as listas de tokens reconhecidos e a dos não reconhecidos. O erro deve estar nos primeiros tokens da segunda lista.  
9- Se o trecho de código estiver correto de acordo com a sintaxe da linguagem ALP, agora é necessário entrar as pré e pós-condições e os invariantes (que são condições, ou seja, expressões booleanas) para a aplicação do cálculo de Hoare. Para cada laço while, será pedido um invariante. Eis aqui o formato das condições (lembre-se que após a expressão booleana deve-se colocar um ponto final para que o interpretador a aceite corretamente):
< cond > ::= true | false | var | ‘Var’ | < cond > and < cond > 

           | < cond > or < cond > | < cond > => < cond > 

           | < cond > <=> < cond > | not < cond > | ( < cond > )
           | ( < Exp > = < Exp > ) | ( < Exp > <> < Exp > )
           | ( < Exp > < < Exp > ) | ( < Exp > > < Exp > )
           | ( < Exp > <= < Exp > ) | ( < Exp > >= < Exp > )
< Exp > ::= num | var | ‘Var’ | - < Exp > | < Exp > + < Exp > 

          | < Exp > - < Exp > | < Exp > * < Exp > 

          | < Exp > / < Exp >

onde var é uma variável começando com letra minúscula, Var é uma variável começando com letra maiúscula (a qual deve vir entre aspas simples para que a condição seja aceita), <=> é a equivalência, => é a implicação e num é um inteiro. Nem sempre os parênteses das comparações são necessários. Devem ser colocados quando a expressão booleana tiver mais termos do que a comparação para evitar ambigüidades.

10- Se uma condição não for entrada corretamente, o programa irá pedi-la novamente. 

11- Agora o programa irá realizar a correção do trecho de código. Assim que o interpretador terminar a execução do programa (quando aparecer a palavra Yes), vá no diretório que o interpretador está usando como local e abra o arquivo saida.html, que possui a saída do programa. 

12- Nele há uma ou mais provas de correção de programa e, para cada uma delas, um conjunto de sentenças a ser demonstradas. Se a pós-condição dada for derivada da pré depois da execução do programa, uma ou mais provas terão todas as sentenças demonstráveis. Escolha como prova aquela que parecer possuir “mais informação” sobre o comportamento do programa. Caso contrário, nenhuma das provas terá todas as suas sentenças demonstráveis.

13- As provas estão em um formato de árvore um pouco diferente do que se vê em livros de lógica. Já o trecho de código é apresentado no formato de árvore de tokens, que é facilmente entendido. Eis aqui a “tradução” do formato de provas:

	formato mais comum
	formato do programa

	{ P } skip { P }
	{ P }  [ skip ] { P }

	{ P[ x / E ] } X := E { P }
	{ P[ x / E ] } [ atribui( X, E ) ] { P }

	{ P } CMD1 { R }  { R } CMD2 { Q }

{ P } CMD1 ; CDM2 { Q }
	{ P } [ CMD1, CMD2 ] { Q }
   { P } [ CMD1 ]{ R }
      ...

   { R } [ CMD2 ] { Q }
      ...     

	{ P ( B } CMD { Q }  P ( ( B ( Q

{ P } if B then CMD fi { Q }
	{ P } [ if( B, CMD ) ] { Q }
   { ( P and B ) } [ CMD ] { Q }
      ...

   ( ( P and not B ) => Q )

	{ P ( B } CMD1 { Q }  { P ( ( B } CMD2 { Q }

{ P } if B then CMD1 else CMD2 fi { Q }
	{ P } [ if( B, CMD1, CMD2 ) ] { Q }
   { ( P and B ) } [ CMD1 ] { Q }
      ...

   { ( P and not B ) } [ CMD2 ] { Q }
      ... 

	_______{ P ( B } CMD { P }_______

{ P } while B do CMD od { P ( ( B }  
	{ P } [ while( B, CMD ) ] { ( P and not B ) }
   { ( P and B ) } [ CMD ] { P } 

      ...

	{ P } CMD { R }  R ( Q

{ P } CMD { Q }
	{ P } [ CMD ] { Q }
   { P } [ CMD ] { R }
      ...

   ( R => Q )

	P ( R  { R } CMD { Q }

{ P } CMD { Q }
	{ P } [ CMD ] { Q }
   ( P => R )

   { R } [ CMD ] { Q }
      ...


Obs: As expressões aparecem em um formato diferente no programa e nas condições e sentenças a serem demonstradas.

6) MINI-PLANO DE TAREFAS:

	data
	duração

(precisão de 15 min)
	descrição da atividade
	esforço

	01/03
	30 min
	determinação da linguagem e sua BNF
	pequeno

	04/03
	15 min
	começar esqueleto do documento final
	pequeno

	09/03
	30 min
	terminar esqueleto do documento final
	pequeno

	09/03
	30 min
	fazer projeto modular inicial do programa 
	pequeno

	10/03
	45 min
	implementar análise léxica
	pequeno

	10/03
	15 min 
	conferir e testar módulo da análise léxica
	pequeno

	10/03
	2 h
	comentar e documentar módulo da análise léxica
	pequeno

	10/03
	1 h
	colocar interface e tratamento de erro no módulo de análise léxica
	médio

	11/03
	1 h e 15 min
	implementar análise sintática
	médio

	11/03
	1 h
	testar análises léxica e sintática
	pequeno

	11/03
	4 h
	procurar interface entre Prolog e ambientes gráficos
	grande

	18/03
	1 h e 15 min
	tentar fazer comunicação entre Prolog e Java   
	grande

	18/03
	1 h e 30 min
	começar a colocar mensagens de erro na análise sintática
	grande

	25/03
	45 min
	corrigir e terminar de colocar mensagens de erro na análise sintática
	médio

	25/03
	30 min
	colocar interface no módulo de análise sintática
	pequeno

	25/03
	2 h e 15 min
	documentar análise sintática
	pequeno

	25/03
	15 min 
	conferir análise sintática
	pequeno

	25/03
	1 h e 30 min
	começar módulo que trata condições
	pequeno

	01/04
	1 h 
	ver como ler e manipular condições
	grande

	01/04
	2 h e 15 min
	corrigir e continuar módulo de condições
	médio

	01/04
	15 min
	escrever resumo do projeto 
	pequeno

	08/04
	45 min
	fazer predicado que verifica se pós-condição vem de pré no axioma de atribuição
	médio

	08/04
	30 min
	continuar módulo de condições
	médio

	08/04
	1 h e 15 min
	colocar interface no módulo de condições e completá-lo
	pequeno

	08/04
	1 h e 30 min
	corrigir predicados do módulo de condições e completá-lo
	médio

	15/04
	45 min
	começar módulo que aplica as regras do cálculo de Hoare (regra do skip)
	médio

	15/04
	15 min
	fazer testes preliminares com o módulo do cálculo de Hoare
	pequeno

	15/04
	2 h e 45 min
	testar, aumentar e melhorar módulo do cálculo de Hoare (regras do skip, atribuição, enfraquecimento de pré-condição e fortalecimento de pós-condição) 
	grande

	15/04
	15 min
	ver problema de resultados espúrios na correção de programas
	grande

	15/04
	30 min
	descrever relações aritméticas 
	médio

	22/04
	15 min
	fazer e documentar módulo de expressões
	pequeno


	data
	duração

(precisão de 15 min)
	descrição da atividade
	esforço

	06/05
	1 h e 45 min
	reestruturar o predicado que aplica as regras do cálculo de Hoare
	grande

	06/05
	1 h
	começar a fazer um predicado que imprime as provas em cálculo de Hoare temporariamente
	grande

	06/05
	30 min
	ver casos teste para o predicado que aplica as regras do cálculo de Hoare
	médio

	06/05
	15 min 
	consertar predicado para o axioma da atribuição no módulo de condições
	médio

	22/04
	30 min
	fazer e documentar módulo principal
	pequeno

	13/05
	1 h
	terminar predicado que imprime as provas em cálculo de Hoare temporariamente
	grande

	13/05
	3 h e 30 min
	continuar o predicado que aplica as regras do cálculo de Hoare (implementar regra para mais de um comando) e testá-lo
	grande

	13/05
	15 min
	colocar retracts no início do programa principal para limpar BD
	pequeno

	13/05
	30 min
	colocar sentenças de lógica válidas e inválidas como um BD na memória principal
	médio

	20/05
	15 min
	corrigir a implementação da regra para mais de um comando 
	grande

	20/05
	1 h
	arrumar as regras de enfraquecimento de pré-condição e fortalecimento de pós-condição
	grande

	20/05
	1 h e 30 min
	colocar mais regras automáticas para a demonstração de sentenças
	médio

	20/05
	1 h
	concluir testes com o que já está feito
	médio

	20/05
	30 min
	implementar as regras do if
	médio

	20/05
	30 min
	ver testes para as regras do if
	pequeno

	27/05
	1 h
	corrigir bugs encontrados
	médio

	27/05
	3 h
	testar regra do if-then e if-then-else
	pequeno

	27/05
	1 h 
	começar a implementar a regra do while
	grande

	27/05
	15 min
	ver saída “menos feia” para as provas
	médio

	06/07
	30 min
	terminar de implementar a regra do while
	médio

	08/07
	1 h e 15 min
	reestrururar programa para não mais fazer perguntas ao usuário e sim mostrar todas as provas e sentenças
	médio

	09/07
	15 min
	impedir enfraquecimentos e fortalecimentos do tipo A -> A
	pequeno

	09/07
	2 h 
	testes
	pequeno

	11/07
	2 h 
	testar regras do if e corrigir seus bugs
	grande

	12/07
	45 min 
	corrigir bugs e começar a testar while
	médio

	13/07
	15 min
	corrigir bug do leitor de loops
	pequeno

	13/07
	1 h
	testar regra do while
	pequeno

	13/07
	30 min 
	fazer interface de saída
	médio

	14/07
	1 h e 15 min
	implementar a construção da página em HTML para a exibição das provas e suas sentenças
	médio

	14/07
	30 min
	fazer filtro para retirar provas 
	médio

	14/07
	15 min
	colocar função que trata erro em arquivo em módulo separado e documentá-lo
	pequeno


	data
	duração

(precisão de 15 min)
	descrição da atividade
	esforço

	14/07
	1 h e 30 min
	começar a documentar módulo que aplica as regras do cálculo de Hoare
	pequeno

	15/07
	45 min
	terminar de documentar módulo que aplica as regras do cálculo de Hoare
	pequeno

	15/07
	15 min
	colocar as referências consultadas na monografia
	pequeno

	15/07
	2 h e 15 min
	colocar atividades já realizadas na monografia
	pequeno

	18/07
	30 min
	colocar objetivos do projeto na monografia
	pequeno

	19/07
	30 min
	colocar requisitos na monografia
	pequeno

	21/07
	15 min
	começar a descrever a interface do programa na monografia
	médio

	22/07
	1 h e 15 min
	corrigir bug na geração do HTML na impressão de condições
	pequeno

	24/07
	15 min
	documentar predicado que imprime condições
	pequeno

	24/07
	15 min
	colocar mais atividades na monografia
	pequeno

	24/07
	30 min
	terminar de descrever interface na monografia
	pequeno

	24/07
	30 min
	descrever funcionamento do programa na monografia
	pequeno

	24/07
	30 min
	descrever projeto modular na monografia
	pequeno

	24/07
	30 min
	colocar o código fonte do módulo lexica.pl na monografia      
	pequeno

	25/07
	1 h e 30 min
	colocar o código fonte dos demais módulos na monografia
	pequeno

	25/07
	15 min
	colocar atividades de ontem na monografia
	pequeno

	26/07  
	30 min
	descrever critérios de testes efetuados
	médio

	26/07  
	30 min
	começar a relatar casos testes
	médio

	27/07
	30 min
	terminar de relatar casos testes
	médio

	27/07
	3 h e 15 min
	começar a fazer testes sistemáticos
	grande

	27/07
	15 min
	colocar atividades de ontem na monografia
	pequeno

	28/07
	3 h e 30 min
	terminar de fazer testes sistemáticos
	grande

	28/07
	30 min
	corrigir bug na regra do while
	médio

	29/07
	2 h 
	fazer a documentação para o usuário
	pequeno

	29/07
	30 min
	conferir monografia
	pequeno


total de tempo gasto no projeto: 85 h

Obs: Tarefas que não produzem resultados (tal como pensar em como fazer algo), não tiveram seu tempo computado nem foram anotadas durante o desenvolvimento do programa. Isto porque, em geral, foram feitas em paralelo com outras atividades que nada tinham a ver com este projeto.

